
Section 4: Visual FoxPro Reference

"But 'glory' doesn't mean 'a nice knockdown argument'," Alice objected.
"When I use a word," Humpty Dumpty said, in rather a scornful tone, "it means just what I choose it to mean—neither more nor less."
"The question is," said Alice, "whether you can make words mean so many different things."
"The question is," said Humpty Dumpty, "which is to be master—that's all."

Lewis Carroll, Through the Looking-Glass, 1872

Section 4 is the meat of the book. You'll find a listing for every command, function, property, event, method and system variable. We've grouped them logically so that you can find several related topics in one place. 

For an explanation of the syntax we use for commands, see "How to Use This Book," back in the Introduction. 

In the printed version of this book, you'll notice that this section is, well, missing. We've had plenty of requests to cut down on the size and weight of this book. Here are a few of those reasons: 

· Conservationists, from those saving the trees to those saving the habitats for animals living in the trees, have expressed a desire for us to reduce the amount of paper used. 

· Workers in the Hentzenwerke stock room and our shipping companies are complaining about back pain and exhaustion from moving copies of the Hacker's Guide around, and they keep muttering things like "worker's compensation." 

· It costs too darn much to ship such a large book, especially to our overseas readers. 

· Nobody wants to purchase it at a conference, because they'd need another suitcase just to bring it home, and it may put them over the airline's cargo limit. 

· We're tired of the jokes about it being a great monitor stand (though the advent of the 17" and larger monitors has drastically cut down on that one), that it makes a great doorstop, and that it provides more exercise than a membership to a fitness center. 

· Nobody reads the paper version anyway; we all use the .CHM version instead. 

So, dear reader, we have listened to your comments, and we've provided this section in an electronic format only. "How do I obtain this wonderful file?" you ask. Very simple. Point your browser to www.hentzenwerke.com. Somewhere on the main page is a link to Downloads. Click that. You'll be given instructions from there on how to get the CHM file. If you purchased this book directly from Hentzenwerke Publishing, you were emailed a user ID and password, and can log in and download the files you want. If you bought the book from another source, you need to have the book in front of you (like now), so you can answer some questions to prove you own it. 

See the section "How to Use the Help File" for lots of useful information, including keyboard shortcuts, how to put the Hacker's Guide into your FoxPro menu, tips on searching, and much, much more. 

Feel free to copy the Help file onto the hard drive of each of your own computers, but please do us the courtesy of not sharing it with everyone you know, even the other folks in your office. (Think of the book as having a single-user license. You will find appropriate copyright notices in the Help file.) We've put a tremendous amount of time into this book, and illegal copies deprive us of the income to pay for that time.

#Define, #UnDef, #If, #ElIf, #Else, #EndIf, #IfDef, #IfNDef, #Include, _Include

Visual FoxPro 3.0 introduced many new concepts to FoxPro programmers, and the ability to manage constants, like the other "real" languages, was a most welcome feature. Visual FoxPro supports the ability to insert defined constants into program, form and class code, as well as the ability to conditionally test and undefine these constants as required. 

These are not FoxPro commands, but rather preprocessor directives. If you include one of these as a string and attempt to macro-expand it within FoxPro, you get the error message "Syntax error." That's because these commands are not intended to be run by the FoxPro interpreter, but are directed to the preprocessor, which produces the pseudo-object code ("p-code"), which the Visual FoxPro interpreter then runs. To distinguish them from FoxPro commands, we often refer to them as "compiler directives," even though, strictly speaking, FoxPro lacks a true compiler. 

Here's the inside scoop. Within the development version of FoxPro, source code can be compiled at various times. When saving a method of a form, the code is compiled during the save process. When running a program whose compiled version is older than its source (with SET DEVELOPMENT ON), the source is recompiled. Finally, source is recompiled when you explicitly tell it to—either by selecting Compile from the Program menu or by building a project. When the preprocessor is called into play, it scans the source code for preprocessor directives, all of which begin with the # symbol. If it finds any of these, it performs the action specified by the command. If this command introduces more preprocessor directives (such as #INCLUDEing another file) these directives are completed, too, until all directives have been performed. At that point, FoxPro source code is converted into the p-code the interpreter (and runtime) will be able to run. Let's review the various commands, and see what they can do. 

	Usage
	#DEFINE cName eExpression

#UNDEF cName


#DEFINE declares a placeholder cName, which is to be replaced at compile time with the expression eExpression defined in the statement. #DEFINE statements work as in-line search-and-replace commands, replacing any occurrence of a specified phrase with an expression. This search and replace occurs in any command line anywhere a command keyword (or even a command itself) can appear. #DEFINE does not affect a literal string set off from the commands with quotation marks, but it does work within square brackets. The example below shows how you might want to take advantage of this. 

#DEFINE is in effect only for the single program that contains the command or the current event or method procedure within a form or class. This is because this is the largest amount of code the preprocessor sees at one time—form methods are each compiled individually. A #DEFINE can be narrowed in scope even further with the addition of an "undefine" (#UNDEF) statement—only code between the #DEFINE and #UNDEF has the substitution performed. 

	Example
	#DEFINE c_nVERSION_NO  "1.23"

* The following inserts the literal TTOC() expression 

* into the source, rather than the date and time.

#DEFINE c_tVERSION_BUILT TTOC(DATETIME())

* 1st example: evaluate both outside of quotes.

WAIT WINDOW "Version "+ c_nVERSION_NO + ;

            ", built "+c_tVERSION_BUILT

* 2nd example: Constants evaluated inside square brackets.

WAIT WINDOW [Version c_nVERSION_NO , built c_tVERSION_BUILT]

* 3rd example: Constant substitution doesn't take place 

* within single or double quotes.

WAIT WINDOW 'Version c_nVERSION_NO , built c_tVERSION_BUILT'


	Usage
	#IF eExpression 

[ #ELIF eExpression ]

[ #ELSE ]

#ENDIF


These statements tests a condition at compile time, and if it's true, the code within the IF...ENDIF pair is included in the source code to be compiled. This lets you include lots of code in your application while developing—tracing and debugging routines, special function key definitions that can pause and single-step the code, routines to provide "backdoors" into certain parts of the application—and prevent the code from even being included in the final version you install at a client site. What a relief not to worry that some data entry clerk will find that "Ctrl+Alt+Shift+F12" routine you used to test the end-of-the-year closeout routine in July! 

#IF directives can also be used to include or exclude platform- or version-specific code. In VFP 3.0 and earlier, slightly different commands or functions are needed on the different platforms to support platform-specific functions. When the new version on the new platform is released, the commands or functions are not recognized on the older software of other platforms, generating compiler errors and defeating the cross-platform compatibility trademark of FoxPro. This new code can be "wrapped" in the #IF...#ELSE…#ENDIF test so that it's included only on the new platform, as the second example below illustrates. 

	Example
	#IF C_DEBUGGING

  do DebugLog with Program(), Pcount()

#ENDIF

 

* Use the faster VARTYPE() if 

* compiling under VFP 6 or later.

 

#IF "FOXPRO 06" $ UPPER(VERSION(1)) OR ;

    "FOXPRO 07" $ UPPER(VERSION(1))  

  IF VARTYPE(toObject) <> "O"

#ELSE

  IF TYPE("toObject") # "O" or ISNULL(toObject)

#ENDIF

    RETURN .F.

  ENDIF


The other directive option used with #IF…#ENDIF is the #ELIF directive. #ELIF, which translates to "Else If," allows you to add multiple decisions to preprocessor directives. Return either a numeric value (0 representing false, a non-zero value representing true) or a logical value to the expression after the #ELIF, to determine whether the successive lines are included. See the example, below. 

	Example
	* 

#IF "FOXPRO 07" $ UPPER(VERSION(1))  

  ? "Using the current version."

#ELIF "FOXPRO 06" $ UPPER(VERSION(1)) 

  ? "Using an older version."

#ELIF "FOXPRO 05" $ UPPER(VERSION(1))

  ? "Using a pretty old version."

#ELSE

  ? "Using a really old version!

#ENDIF


	Usage
	#IFDEF cConstantName

#ENDIF


This pair of statements tests for the existence of a predefined constant, declared with the #DEFINE statement above, at compile time, and if it's defined, the code within the IFDEF...ENDIF pair is included in the source code to be compiled. 

	Example
	#IFDEF cAuditing  && Auditing functions to be included

  IF cAuditing

    DO AuditTrail with ...

  ELSE

.... more code

#ENDIF


	Usage
	#IFNDEF cConstantName

#ENDIF


This pair of statements is the reverse of the above, the equivalent of NOT in the IF portion of the test. If a predefined constant is not in effect, either through the lack of a #DEFINE statement or the removal of the definition using an #UNDEF, the code within the IFNDEF...ENDIF pair is included in the source code to be compiled. 

	Example
	#IFNDEF C_DEBUGGING

  SET DEVELOPMENT OFF

  SET TRBETWEEN OFF

#ENDIF


	Usage
	#INCLUDE cFileName

_INCLUDE = eFileAndPathName


#INCLUDE inserts a second file into the first file, in memory, just before performing the compile. This allows you to keep a set of handy #DEFINEs around, written in just one place, and stick them into all of your programs. A couple of caveats apply, of course: 

One very neat thing is that a file that's #INCLUDEd can have any or all of the preprocessor directive statements within it, and these, too, are processed by the preprocessor. Our suggestion: Break constant files into bite-sized chunks and create a STANDARD.H file that #INCLUDEs them all. That way, if a specific set of definitions must be maintained, they can be located most easily. Breaking them up by function (display, I/O, security) can make it easier to swap out a set when replacing or upgrading a subsystem. An alternative is to have a "common" set of definitions in a COMMON.H and include that in "subsystem" include files. 

Rather than using this function in every method of a form or class, check out the "Include File..." option on the Form or Class menu. This makes the #INCLUDEd material available to each and every method within the form or class. 

	[image: image1]
	In VFP version 3.x, compiling a file that #INCLUDE's other files doesn't return any error messages if the #INCLUDEd file is not located or cannot be opened because it is in use. This one trips us up all the time: If you open FoxPro.H to see what they called a constant, and then compile and run a program using that file without closing the editing window, you'll get stupid errors like "Variable 'COLOR_RED' is not found." In VFP 5.0 and later, a dialog appears, informing you the file can't be opened, and generates an ERR file if you choose "Ignore" to continue compiling. Although we at least get an error telling us it's open, we wish having the #INCLUDE file open didn't prevent you from compiling. That's just plain annoying. 


	[image: image2]
	We've mentioned FoxPro.H in a few places throughout the book, and thought we should clue you in on this file. This file contains predefined sets of constants for many of the more difficult-to-remember functions, such as color numbers, low-level file routines, PRTINFO() types, parameters to MessageBox() and so forth. This file is worth browsing and using. We far prefer to see code using these constants, because it's much easier to read and, therefore, to maintain.



_INCLUDE is a system memory variable introduced in VFP 6. _INCLUDE is blank unless we set the preferences in Tools | Options | File Locations. When _INCLUDE is set to a header file, this file is automatically included in the compilation of all forms and classes. This is a wonderful way to ensure that your constants are available throughout your projects. We wish, however, they would have considered making _INCLUDE available for all compilation—programs, menu code, whatever—for consistency. 

While .H is a standard extension for a header file, you can use any extension you like, provided you specify it in the #INCLUDE statement. Do remember that other programmers are likely looking for the .H extension, though. 

	Example
	* STANDARD.H

#INCLUDE FOXPRO.H

#INCLUDE MyVars.H

 

* Much easier-to-read code

#INCLUDE FOXPRO.H

IF MESSAGEBOX("Question",;

              MB_YESNO + MB_ICONQUESTION + MB_DEFBUTTON2, ;

              "Title") = IDYES

* than:

IF MESSAGEBOX("Question", 292, "Title") = 6  && what's 292? 6?

 

_INCLUDE = LOCFILE(HOME()+"FOXPRO.H","H")


	See Also
	Compile, Set Development


At(), AtC(), RAt(), $

Did you ever need to know whether a string contains a particular substring? Say you have a memo field containing notes about a phone conversation and you want to see whether you discussed the "Super Duper Pooper Scooper." Enter this group of functions. 

Technically, $ is an operator, while AT(), ATC() and RAT() are functions. But all of them are used to find a substring within a string. $ simply indicates whether or not the substring is there. The others all return the position where the string was found. 

	Usage
	lIsItInThere = cSearchFor $ cSearchIn

nFoundPos = AT( cSearchFor, cSearchIn [ ,nOccurrence ])

nFoundPos = ATC( cSearchFor, cSearchIn [ ,nOccurrence ])

nFoundPos = RAT( cSearchFor, cSearchIn [ ,nOccurrence ])


AT() and RAT() are case-sensitive; ATC() is not (the "C" stands for "case," which seems backward to us, but at least the function exists). AT() and ATC() begin their search at the left-hand side of the string, while RAT() begins at the right (you can read it as "Right AT"). 

Watch out for one thing with RAT(). Even though it searches from the right, the position it returns is measured from the left. We guess that's the "RAT trap." 

All three functions take an optional parameter indicating which instance of the string to search for, so you can find, say, the third use of "Super Duper Pooper Scooper" rather than the first or last. 

You can use these functions together with LEFT(), RIGHT() and SUBSTR() to parse a string into its component parts. This is often useful when you inherit data that hasn't been properly normalized. For example, you might need to convert a single name field into first and last name fields or to pull apart an address field into street address, city, state and zip code. 

Be aware that these functions can't be optimized by Rushmore (unless you have some pretty unusual and not terribly useful index tags), so you won't want to use them heavily for lookups. For example, don't store several codes in a single field and plan to use $ to see if a specific code is present; normalize the data instead. 

	Example
	lFoundIt = "Super Duper Pooper Scooper" $ mNotes

nStartPos = AT("Super Duper Pooper Scooper", mNotes)

nStartPos = ATC("Super Duper Pooper Scooper", mNotes)

nLastOne = RAT("Super Duper Pooper Scooper", mNotes)

 

* Here's a more useful example which shows how you'd take apart 

* a field containing city, state and zip code to create separate 

* fields. You'd call this routine like this:

STORE "" TO cCity, cState, cZip

DO PARSADDR WITH cCityStZip, cCity, cState, cZip

 

* parsaddr.prg

* Parse single address variable into city, state and zip

* Assumes parameter cAddress has structure:

*    City, ST Zip

* Zip can be either 5 or 10 digit

 

LPARAMETERS cAddress, cCity, cState, cZip

 

LOCAL nStartZip, nStartState

 

nStartZip = RAT(" ",TRIM(cAddress))

cZip = TRIM(SUBSTR(cAddress, nStartZip+1))

 

nStartState = AT(",", cAddress)

cState = ALLTRIM(SUBSTR(cAddress, nStartState+1, ;

                 nStartZip-nStartState-1))

 

cCity = ALLTRIM(LEFT(cAddress, nStartState-1))

 

RETURN


VFP 3.0b added double-byte versions of the functions in this group: At_C(), AtCC(), RAtC(). They're the ones to use when you're working in a language with double-byte characters or if there's a chance your application will go international. 

	See Also
	AllTrim(), AtLine(), AtCLine(), At_C(), AtCC(), Left(), Occurs(), RatC(), RatLine(), Right(), Substr(), Trim(), Upper()


 

%, Mod()

The % operator and MOD() both compute the modulus of a pair of numbers. For those who never liked math in the first place, that's the remainder when you divide the first number by the second. (Actually, it is for the rest of us, too.) Another way to think of it is as the part you throw away when you apply FLOOR() to the quotient. 

	Usage
	nRemainder = nDividend % nDivisor

nRemainder = MOD( nDividend, nDivisor )


	Parameter
	Value
	Meaning

	nDividend
	Currency, Double, Integer, Numeric
	The number being divided. (The number after "goes into.")

	nDivisor
	Currency, Double, Integer, Numeric
	The number doing the dividing. (The number before "goes into.")

	nRemainder
	Number
	The remainder when nDividend is divided by nDivisor.



MOD() and % are pretty straightforward when dealing with positive numbers, but they get interesting when one or both of the numbers is negative. The key to understanding the results is the following equation: 

MOD(x,y) = x - (y * FLOOR(x/y))

Since the mathematical modulo operation isn't defined for negative numbers, it's a pleasure to see that the FoxPro definitions are mathematically consistent. However, they may be different from what you'd initially expect, so you may want to check for negative divisors or dividends. 

A little testing (and the manuals) tells us that a positive divisor gives a positive result while a negative divisor gives a negative result. 

MOD() is most useful when you want to set up intervals of some sort. For example, you might use it in a DynamicBackColor condition to get alternating colors in a grid. 

	Example
	? 3%10         && Returns 1

? MOD(22, 10)  && Returns 2

? MOD(-7, 3)   && Returns 2

? MOD(-7, -3)  && Returns -1

? MOD(7, -3)   && Returns -2


	[image: image3]
	The type of the result depends on the divisor and the dividend. When you mix Currency and Numeric or Integer, the result is always Currency. But when you mix Currency with Double, the divisor determines the result type, which is as it should be. The behavior of Currency with numbers feels like a bug to us. In any case, the various numeric types should all behave the same way.


	See Also
	Ceiling(), Floor()


(), Evaluate(), &, ExecScript()

Sometimes you need to refer to a file without knowing its name or you need to build an expression or even a series of lines of code and then execute it. Think of () (called "indirect reference" or "name expression"), EVALUATE() (better known as EVAL()), & (macro expansion), and EXECSCRIPT() as the guillotine that performs the execution. Hmmm, maybe that's not such a good analogy. 

In any case, all four items let you write code when you don't know exactly what should happen when the code runs. There are four levels, which correspond to the four choices. 

	Usage
	(cName)

uResult = EVALUATE(cExpression)

&cString

uResult = EXECSCRIPT(cCode)


First, perhaps you want to let a user select a report to run, or you'd like to use the same piece of code to open several different tables. Indirect reference lets you put a name in a variable and refer to the variable instead of giving the specific name. For example, if you write: 

USE (m.cFileName)

FoxPro evaluates the variable cFileName and substitutes the name found there in the USE command. 

Indirect reference works only when FoxPro expects a name. So you can't write something like: 

BROWSE FOR (m.cExpression)

because FoxPro expects an expression there, not a name. 

Which brings us neatly to the next level. When FoxPro expects an expression, you can use EVALUATE(), which tells FoxPro to evaluate the named variable, and then use the result as part of the command. (Incidentally, TYPE() and EVALUATE() use exactly the same level of indirection. If you have trouble figuring out what parameter to pass to TYPE(), imagine that you're passing it to EVALUATE() to be evaluated.) 

For example, in the BROWSE above, you could write: 

BROWSE FOR EVAL(m.cExpression)

It turns out, though, that it's not a great idea to use EVALUATE() in a FOR clause. We'll explain why below. But here's a better example of EVALUATE(): 

STORE EVAL(cFieldName) TO SomeVar

EVALUATE() is essential in reports, where macro expansion doesn't work. You can use expressions like: 

EVAL(FIELD(1))

to handle reporting on query results where the field names are unknown (such as in a crosstab). You can also use: 

EVAL(cGroupExpr)

as a grouping expression to avoid designing separate reports where grouping is the only difference. 

The third level is for the times when you want to build up whole pieces of code and then execute them. In this case, the string contains more than just an expression—perhaps even an entire command. Here, you need macro expansion. (These are "real" macros, not to be confused with the wimpy keyboard macros covered in SAVE MACROS, PLAY MACRO, and RESTORE MACROS. Keyboard macros are just a simplistic recording and playback of keystrokes. This macro expansion—arguably one of Fox's most powerful commands—tells Fox to interpret, compile and execute the code on the fly.) Macros are also used to substitute for reserved words. For example, you've probably written something like this: 

cOldSafety=SET("SAFETY")

SET SAFETY OFF

* some code

 

SET SAFETY &cOldSafety

Here, the "ON" or "OFF" that was stored in cOldSafety is a character string, but the command SET SAFETY expects a keyword of ON or OFF. The & macro operator converts the string to a keyword. 

Another common place for a macro is in queries. You might build up any of the field list, the table list or the WHERE clause in a string and then macro-expand it like this: 

SELECT &cFieldList ;

    FROM &cTableList ;

    WHERE &cWhere

Macros are powerful and work almost everywhere (though not, as noted above, in reports). There are a couple of gotchas, though. 

First, only variables can be macro expanded. You can't apply a macro to a field or property. So, you need two steps when the string to be expanded is in a field or property. Say the field cCmd of MyTable contains an entire command to be executed by macro expansion. You have to do it like this: 

cVar=MyTable.cCmd

&cVar

The second gotcha is sort of a consequence of the first. When you macro-expand a variable, you can't use the "m." variable notation. This is okay because you can't macro-expand a field, so there's no ambiguity. But why does it work this way? 

The "." is a terminator for macro expansion. (Boy, this is a gory topic. We've got "executed" commands and "terminators." Where's Arnold?) This means the string to be expanded ends as soon as you reach a period. This can be very convenient if you want to add something to the expanded string or expand several strings in sequence. Before the addition of indirect reference to the Fox language in 2.0, it was essential because you could write something like: 

USE &cFileName..EXT

but now, with named expressions, it's better to write: 

USE (cFileName+".EXT")

We still find uses for the double period in something like: 

lcAlias = alias()

* more code in here

xx = &lcAlias..Field1

or: 

ThisForm.pgfPageFrame.&PageName..Control.Property = .T.

though you could use EVAL(lcAlias+".Field1") in the first case. 

The consequence of the period as terminator is what happens when you write something like: 

&m.cVar

FoxPro sees this as "macro expand m, then tack the results onto 'cVar'." Since you don't usually have a character variable called "m", generally you get an error message when you do this. Even if there is a variable m, you certainly don't get what you wanted. 

What about performance? We often see people going to great lengths to eliminate macros from their code. But the truth is that, in certain cases, macros are actually faster than EVALUATE(). Here are a couple of guidelines. 

Always use indirect reference rather than EVALUATE() or a macro. If FoxPro expects a name, indirect reference is the right choice. 

In a command that will be executed repeatedly, it's generally better to use a macro than EVALUATE(). The macro is expanded once and substituted into the command while an EVALUATE() expression is evaluated each time the command executes. This is also true for scoped commands that apply to multiple records, like the BROWSE above. The macro is expanded once and the result used on each record; EVALUATE() is re-evaluated for each record. So, for the BROWSE above, you're better off issuing: 

BROWSE FOR &cExpression

Macros that execute only once aren't really all that slow. The place to avoid macros is inside loops. 

Indirect referencing, EVALUATE(), and macros fill in parts of a line of code, or can even stand in for a whole line of code. But what if you want to compose and run multiple lines of code? In versions prior to VFP 7, you had to use a utility like Randy Pearson's CodeBlock or RunCode.PRG from the VFP FFC subdirectory. But VFP 7 gives us EXECSCRIPT(), which runs many lines of code, much like highlighting several lines in the Command Window and executing them (they're probably running the same code internally). EXECSCRIPT() takes a character string as its first parameter, which can be a variable, character field, or memo field. Subsequent parameters are passed to the code contained in the first parameter. 

Perhaps you have a table of user-defined formulas. They all take two numeric parameters, nFirst and nSecond. A really cool module of your application allows them to define the formula, then stores it in a memo field in the UDCode table. The memo field, cleverly called cCode, might contain: 

LPARAMETERS nFirst, nSecond

RETURN nFirst + nSecond

To execute the code in the memo field, issue: 

nNewValue = EXECSCRIPT(UDCode.cCode, 2, 4)

Since EXECSCRIPT() returns a value, you can embed it in other statements, such as within SELECT commands. The possibilities are mind-boggling. You can store code in memo fields, perhaps providing custom functionality for individual clients. You can build and execute multiple lines on the fly, built from information entered by the user. 

	See Also
	Play Macro, Restore Macros, Save Macros, Type()


DoDefault(), ::

This function and the operator composed of two colons (called the "Scope Resolution" operator by the docs) let you call methods higher up in the class hierarchy. You use them most often when you want to augment a method to do everything the parent class's method does, plus some more. 

	Usage
	uReturn = DoDefault( [ uParmList ] )

uReturn = ClassName::Method( [ uParmList ] )


	Parameter
	Value
	Meaning

	uParmList
	List of expressions
	Parameters to pass to the method being called.

	ClassName
	Name
	The name of the class whose method you want to call.

	Method
	Name
	The name of the method to be called.

	uReturn
	 
	The value returned by the method called.



DoDefault(), added in VFP 5, calls directly up the class hierarchy. That is, it can call only the parent class's version of the method you're in. The :: operator gives you more flexibility than that, but most often, you use it the same way. You can, in fact, call any method that's in the class's inheritance tree. You can't call a method that belongs to a class the current class doesn't inherit from. 

Despite the fact that :: is more flexible, it's better to use DoDefault() when you can because it doesn't tie your code to a particular class or method name. Your code is more reusable if you allow the DoDefault() to find the appropriate class hierarchy for the method it finds itself in. 

	[image: image4]
	The initial release of VFP 5 had a bug in DoDefault() that caused a problem when there was a method in the hierarchy that didn't have any code. If you issued DoDefault() in a method and the same method in the parent class was empty, the method in the parent class's parent class executed twice. The bug was fixed in VFP 5.0a.


	[image: image5]
	VFP 5 and early versions of VFP 6 have a truly insidious bug as well. It occurs if you use DoDefault() in a method that automatically calls the same method of other objects (for example, a form Refresh, which automatically calls Refresh of member objects, or KeyPress when KeyPreview is .T., which automatically forwards the keypress to the object with focus) and there's no code in the same method of the parent class of the original object (the form, in the examples). In that case, the keyword This in the other methods of the contained object(s) (the member objects in the Refresh example, the object with focus in the KeyPress example) doesn't refer to the object whose code is being executed, but to the original object whose code had the DoDefault() (the form, in the examples). This was a truly ugly bug and we glad it got exterminated.


	Example
	DEFINE CLASS CloseButton AS CommandButton

   * Set appropriate properties up here

   * including

   Caption = "Close"

 

   PROCEDURE Click

      ThisForm.Release

   ENDPROC

 

ENDDEFINE

 

DEFINE CLASS ConfirmCloseButton AS CloseButton

 

   PROCEDURE Click

      LOCAL nResult

      nResult = MESSAGEBOX("Closing Form",33)

      IF nResult = 1    && OK

         DoDefault()

         * Or, in VFP 3

         * CloseButton::Click

      ENDIF

   ENDPROC

 

ENDDEFINE


	See Also
	Class, Parent, ParentClass, This


=, Store

This operator and command let you assign a value to a memory variable. The results are the same, but STORE lets you save the same value to multiple variables. Both let you save a single value to all elements of an existing array. 

	Usage
	Variable = uExpression

STORE uExpression TO Variable1 [, Variable2 [...] ]


STORE and = work only on variables and properties. A very common mistake (we can't tell you how often we've made it) is to try to use = to store a value in a field; it doesn't work. In fact, unlike most places in FoxPro, if you have a memvar and field with the same name, STORE and = assume you mean the memvar. (Most commands assume an unqualified, ambiguous reference is to a field.) Because of this, there's never a reason to use the "m." notation when assigning a value with = or STORE, and there's good reason not to—it turns out that assignment is faster without the "m." 

Another way to save time with STORE is to use a single statement to initialize a bunch of variables: 

STORE 0 to nOne, nTwo, nThree, nFour, nFive

is faster than: 

nOne = 0

nTwo = 0

nThree = 0

nFour = 0

nFive = 0

STORE is also handy when you're writing generic code because it lets you use a name expression rather than a macro, typically a faster and less resource-intensive operation. That is, if cVarName contains the name of a variable, you can write: 

STORE 0 TO (cVarName)

rather than: 

&cVarName = 0

If the variable is an existing array and SET COMPATIBLE is OFF, the value of uExpression is stored in every element of the array. This is a quick way to initialize an array. With SET COMPATIBLE ON, the array is overwritten by a single memvar that gets the value of uExpression. 

Objects change the game with = and STORE. You can use them with objects. What you get is a new reference to the same object, not a copy of the object. There isn't any easy way to create an exact copy of an object. We can think of some brute-force ways, but they're not pretty. 

	Example
	dToday=DATE()

STORE 0 TO nTotal,nCount

DIMENSION aTotals[5]

aTotals=5

oObj1.SomeProperty="Old Value"

oObj2=oObj1

oObj1.SomeProperty="New Value"

? oObj2.SomeProperty          && Returns "New Value"


	See Also
	Replace, Set Compatible


?, ??, Set Space, Set("Space")

These commands let you produce streaming output (as opposed to the line-oriented output of @ .. SAY). ? and ?? produce output. SET SPACE determines whether a space is used between output items. 

	Usage
	? | ?? [ uExpr1 [ PICTURE cMask ] 

              [ FUNCTION [ cCodes ] [ V nWidth ] ]

       [ AT nColumn ] 

       [ FONT cFontName [, nFontSize ] [ STYLE cStyleCodes ] ]

       [ , eExpr2 [ ... ] ] ]

SET SPACE ON | OFF

cSpaceSetting = SET("SPACE")


	Parameter
	Value
	Meaning

	uExprn
	Expression
	The nth expression to print.

	cMask
	Character
	Output mask using the same codes as the InputMask property.

	cCodes
	Character
	Function codes using the same list as the Format property.

	nWidth
	Numeric
	The number of columns to devote to this item. 

	nColumn
	Numeric
	Starting column for this item.

	cFontName
	Character
	Font to use for this item.

	nFontSize
	Numeric
	Font size to use for this item.

	cStyleCodes
	Character
	List of style codes to use in the specified font.



? streams output including a carriage return and line feed, while ?? sends output without those characters. When we first learned Xbase, it took us the longest time to understand that ? sends the CR/LF pair before it prints. So, to send several items on one line followed by a new line, you use a series of ?? and then issue ?. 

Over the years, ? and ?? have gotten more and more powerful, acquiring much of the functionality of @ .. SAY. We use these a lot more than @ .. SAY, though, because they're handy for sending output to the active window when you're testing. However, we rarely use either one in applications. 

When you include several expressions in a single command, each can have its own font and position clauses. Be careful how you combine them, though, because the column you specify in the AT clause is computed based on that item's font. It's easy to overwrite one item with another. 

??CHR(7) is the traditional way to sound the bell in FoxPro. It still works just fine with one warning. You'll probably want to specify a small font for it because it can (eventually) make the active window scroll up. Two-point Arial works just fine. (SET BELL controls the frequency and duration of the tone.) 

By default, when you include multiple items with ? or ??, a single space appears between them. SET SPACE OFF to run them together. 

	Example
	? $12345 PICTURE "999,999,999" FUNCTION "$"  && displays $12,345

USE Employee

? First_Name,Last_Name   && Looks good

SET SPACE OFF

? First_Name,Last_Name   && Looks awful


	See Also
	@...Say, Format, InputMask, Set, Set Bell, Set Print


Set TextMerge, Set("TextMerge"), \, \\, _PreText, _Text, Set TextMerge Delimiters, TextMerge()

Textmerge is a combination of several of FoxPro's best features—it creates low-level file output, performs runtime evaluation and eases formatting of different types of data. \ and \\ are the textmerge equivalents of ? and ??, outputting to the destination specified by SET TEXTMERGE TO, rather than SET DEVICE TO. Starting in VFP 7, you have several new options for creating textmerge output, including the ability to send the results to a variable. 

	Usage
	SET TEXTMERGE [ ON | OFF ] | 

        [ TO FileName | TO MEMVAR VarName [ ADDITIVE ] ]

        [ WINDOW WindowName ]

        [ SHOW | NOSHOW ]

cOnOrOff = SET( "TEXTMERGE" )

cMergeFile = SET( "TEXTMERGE", 2)

cShowOrNot = SET( "TEXTMERGE", 3)

nRecursionLevel = SET( "TEXTMERGE", 4)


Here's the deal. When you issue the command SET TEXTMERGE TO FileName, a low-level file channel is opened and the file FileName is opened or created. The low-level file handle is stored in the system memory variable _TEXT. When you use the TO MEMVAR clause (added in VFP 7), there's no file involved, but the named variable is created, if necessary. Output created with the single or double backslash (\ or \\) is echoed to the file or stored in the variable, with \ issuing a carriage return before outputting the remainder of the line, and \\ issuing its characters immediately following the text that had been output before. "Big deal," you say. "I can do the same thing with SET ALTERNATE or SET PRINTER." Yes, BUT! Textmerge really shines when TEXTMERGE is set ON. 

SET TEXTMERGE ON tells FoxPro to examine each line of output for expressions encased in textmerge delimiters. If these are found, the expression within delimiters is evaluated, and the result of that expression is output. Still not impressed? Here's the key: The result of the expressions is converted to text automatically—dates, numerics, datetimes, whatever. To output a line containing a number, date, datetime and page number, you would have to convert each one, as in: 

?  LTRIM(STR(nNumber,4)) + space(6) + ;

   DTOC(Date()) + SPACE(7) + ;

   TTOC(DateTime()) + SPACE(4) + ;

   "Page #" + ltrim(_PAGENO)

In a textmerge document, you would just say: 

\<<nNumber>>      <<Date()>>       <<Datetime()>> Page # <<_PAGENO>>

Now which would you prefer to have to decode and maintain six months after you wrote it? 

	[image: image6]
	SET TEXTMERGE is different from SET ALTERNATE or SET PRINT in that you can both SET TEXTMERGE TO and ON in a single line. For ease of maintenance, we advise you to splurge on the two lines of code to SET TEXTMERGE TO and ON.


	[image: image7]
	Unlike the command to SET TEXTMERGE ON and SET TEXTMERGE TO, you must write two separate lines of code to SET TEXTMERGE OFF and SET TEXTMERGE TO, so that's an even a better reason to have two matching pairs of commands.



And textmerge is fast! Ted wrote some processing code a while back to read through 800 HTML documents, parse the contents, and generate an HTML Help index in SiteMap format. Processing all of the files took eight seconds! One key factor was to set NOSHOW. With the text scrolling past on the screen, it took eight minutes for the same process. 

	[image: image8]
	The original release of VFP 7 sometimes crashes when you mix TEXT TO processing with SET TEXTMERGE processing. It's fixed in SP1.



The remaining optional clauses are pretty straightforward. ADDITIVE specifies that output is appended to the end of an existing file or variable. WINDOW WindowName allows you to specify an output window where the echoed textmerge text should appear. We recommend you always specify a window if you want the output echoed, for two reasons. First, in Visual FoxPro 5 and later, it's possible to release the main FoxPro window and run your application as a top-level form. In that case, your output might be lost. Secondly, it's been our experience that Windows seems to slow down when outputting scrolling text to the screen or a window not on top. If you create a specific window for your textmerge, you can force it to be WONTOP() for the operation. 

SHOW | NOSHOW specifies whether text output to the merge file should also be echoed to the screen or optional window. 

	[image: image9]
	The help file claims that SHOW is the default—that's true the first time SET TEXTMERGE is issued. After that, the default appears to be the last setting used—that is, if NOSHOW was issued with the previous command, text will not be echoed unless SHOW is explicitly specified. So this isn't really a "default" behavior, it's more like a global SET behavior. Always specify SHOW or NOSHOW and you won't have a problem.



If you choose to send output to a file, rather than using SET TEXTMERGE TO a particular file, we suggest you use FCREATE() or FOPEN() to access the file. Why? If SAFETY is ON, SET TEXTMERGE generates a confirmation dialog to overwrite the file. Any number of errors can prevent the file from being created—improper file names, bad drive designator, lack of rights—but FCREATE() and FOPEN() allow you to handle the errors using the simpler low-level error handler FERROR(), rather than the massive CASE statement needed to process all of FoxPro's possible file errors, or dropping though to your global error handler. If the low-level function was successful, you can set _TEXT to the file handle returned from these low-level file functions. 

Overall, though, we recommend sending textmerge output to a string, then if a file is needed, using StrToFile(). 

There are other neat commands to use with text merging. TEXT...ENDTEXT allows you to include a block of text right in your program; it obeys the settings of _PRETEXT and evaluates any expressions within delimiters. It's also been enhanced in VFP 7 to include a lot of its settings right in the command. 

The various forms of SET("TEXTMERGE") let you see how you have things set up. Some of them work better than others and some of them make more sense than others. 

SET("TEXTMERGE"), by itself, returns either "ON" or "OFF". SET("TEXTMERGE",2) gives you the name of the file to which you've SET TEXTMERGE. SET("TEXTMERGE",3) tells you whether textmerge is currently SHOW or NOSHOW. Since textmerge can be recursive (when the string merged in contains the textmerge delimiters), SET("TEXTMERGE", 4) tells you how many levels of recursion are pending at the moment. 

	[image: image10]
	Unfortunately, SET("TEXTMERGE",4) uses a ridiculously narrow definition of recursion. It adds a level to the count only if the expanded string contains the same variable from which it was expanded. That is, if you expand the variable ExpandMe and its text contains the string "<<ExpandMe>>", the count goes up (which it should). However, when talking about textmerge, the term "recusion" generally applies to any expanded string that contains the textmerge delimiters, not just expanding the same string over and over. SET("TEXTMERGE",4) doesn't go up if you expand the variable ExpandMe and it contains the string "<<MeToo>>".

	There's no way to find out what variable you've SET TEXTMERGE to. 


	Example
	* The *best* examples of textmerge are available in your main 

* Visual FoxPro directory - check out GENMENU.PRG and others.

* But here's something for you to try:

LOCAL lcOldPreText

* SET TEXTMERGE ON TO textmerg.txt NOSHOW  && Don't do this!

SET TEXTMERGE TO textmerg.txt

SET TEXTMERGE ON NOSHOW

\Generated at <<DATE()>> <<TIME()>>

\

TEXT

This is a block of text with

no comments or anything before 

it

ENDTEXT

\

\Now we add a pretext

\

lcOldPreText = _PRETEXT

_PRETEXT = "*"+CHR(09)

TEXT

Four score and seven years ago,

our forefathers (and mothers) brought

forth upon this continent a new nation,

ENDTEXT

_PRETEXT = lcOldPreText

\

\ And delimiters are evaluated within TEXT...ENDTEXT

\

TEXT

Today is <<DATE()>> and the time is <<TIME()>>

ENDTEXT

SET TEXTMERGE OFF 

SET TEXTMERGE TO

MODI FILE textmerg.txt 

 

* Textmerge to a variable

SET TEXTMERGE TO MEMVAR cHTML

SET TEXTMERGE ON NOSHOW

\<HTML>

\<HEAD>

\<TITLE><<Customer.Company>></TITLE>

\</HEAD>

\<BODY>

\Address: <<Customer.Address>>

\</BODY>

\</HTML>

* and so forth to build an HTML document

SET TEXTMERGE OFF

SET TEXTMERGE TO

StrToFile( cHTML, "MyHTMLDoc.HTM" )


	Usage
	_PRETEXT = cTextToPrecede

cTextToPrecede = _PRETEXT

_TEXT = nHandle

nHandle = _TEXT


_PRETEXT is not a system-generated excuse like "I just happened to be in the neighborhood," but a system memory variable that holds a character expression. This expression is inserted at the beginning of each line of text generated by the textmerge commands. Set _PRETEXT to "*" to comment a block of code, or to a set of tabs to indent a block. 

_TEXT is a built-in FoxPro system memory variable. It contains the low-level file handle of the destination of textmerged output. Set _TEXT to –1 (negative one) to temporarily shut off output to the designated file. Since _TEXT stores the number of a low-level file handle opened for output, switching _TEXT programmatically allows you to switch output back and forth between several destinations, a trick the FoxPro 2.x screen generator used to separate startup and cleanup code. 

_TEXT is set to –1 when you SET TEXTMERGE TO a variable. More importantly, the current textmerge output file is closed at that time, so you can't save the value of _TEXT and restore it after sending textmerge output to a variable. 

	Example
	_PRETEXT = "*" + CHR(9)  && Comment and indent merged text.

SET TEXTMERGE TO D:\TEXTMERG.TXT

? _TEXT  && Displays the opened file handle.


	Usage
	SET TEXTMERGE DELIMITERS TO [ cLeftExp [, cRightExp ] ]

cAllDelimiters = SET( "TEXTMERGE", 1 )


Delimiters specified with this command tell FoxPro what to look for when outputting a line with the textmerge commands \, \\, TEXT or TEXTMERGE(). Expressions contained within the delimiters are evaluated when TEXTMERGE is SET ON. If no delimiters are specified, they default to << and >>, respectively. If a single character expression is specified, it's used for both the left and right delimiters. Avoid using those single characters already used by other FoxPro functions—%, &, $, *, (, or ). Colons, used singly or doubly, are probably a safe bet, as long as you don't use the scope resolution operator in the code. It's best to leave the delimiters as is, unless you need to output the << or >> characters themselves. Each delimiter expression can be either one or two characters in length. 

	[image: image12]
	This means that you can't really parse the return value of the SET() function and be sure where the first delimiter ends and the second begins—see the second example below. So a "black box" routine cannot mess with these values. A return length for SET("TEXT",1) of 3 means you're doomed, since you don't know which one of the delimiters has two characters. It would be better if this SET() function, like most of them, returned the explicit string that we could use to SET TEXTMERGE DELIMITERS TO &OurString, and leave the parsing to us. Our advice: Don't do this. Leave the delimiters as they are, or if you must change them, change them to something simple like a matched pair of single or double curly braces or something.


	Example
	SET TEXTMERGE DELIMITERS TO "::", "::"

 

SET TEXTMERGE DELIMITERS TO "@~","@"

? SET("TEXTMERGE",1)  && returns "@~@" - but which is which?


	Usage
	cMergedOutput = TEXTMERGE( cString [, lRecursive

                           [, cLeftDelim [, cRightDelim ] ] ] )


	Parameter
	Value
	Meaning

	cString
	Character
	The string to be evaluated.

	lRecursive
	.T.
	If a merged expression contains the textmerge delimiters, keep evaluating the contents of the delimiters again until no more delimiters are found.

	
	.F. or omitted
	Evaluate strings inside the textmerge delimiters only once.

	cLeftDelim
	One or two characters
	The left delimiter for textmerge within the function.

	
	Omitted
	Use the current left delimiter from the SET TEXTMERGE DELIMITERS setting.

	cRightDelim
	One or two characters
	The right delimiter for textmerge within the function.

	
	Omitted
	Use the current right delimiter from the SET TEXTMERGE DELIMITERS setting.



This function, added in VFP 7, makes textmerge easier than ever. No need to issue SET commands; no need to put text inside a TEXT … ENDTEXT block. Just build the string and call the function. 

If you don't specify delimiters in the function call, the current textmerge delimiters are used. However, specifying delimiters in the call doesn't change the current delimiters. Unlike SET TEXTMERGE DELIMITERS, with TEXTMERGE(), you can specify a left delimiter without specifying a right delimiter. In that case, the left delimiter in the function call is used, along with the current right delimiter. 

	Example
	#DEFINE CRLF CHR(13) + CHR(10)

cString = "<HTML>" + CRLF

cString = cString + "<HEAD>" + CRLF

cString = cString + "<TITLE><<Customer.Company>>" + ;

    "</TITLE>" + CRLF

cString = cString + "<BODY>" + CRLF

cString = cString + "Address: <<Customer.Address>>" + CRLF

cString = cString + "</BODY>" + CRLF

cString = cString + "</HEAD>" + CRLF

cString = cString + "</HTML>"

cHTML = TextMerge( cString )


	See Also
	FError(), Low-Level File Functions, Set Alternate, Set Printer, Text ... EndText


@ Commands

These commands were among the most important commands in every version of FoxBase and FoxPro until VFP. With the movement to control-based forms, the @ commands for controlling input and output move over to the "included for backward compatibility" category. 

Don't write any new code in Visual FoxPro that uses the @ commands. As for converting older code, it's a hard call. So far, we're leaving our existing apps happily in 2.x and beginning only new projects or complete revisions in Visual FoxPro. 

Nonetheless, you might have to move some old screens into Visual FoxPro. You have two choices: Maintain the screen in 2.x and use the generated SPR in Visual FoxPro, or let the Converter start the transition toward object-oriented input. The Converter itself also offers two choices—Functional Conversion and Visual Conversion. Functional Conversion gives you something you can use immediately, but at the cost of pretty ugly internal structure. Visual Conversion leaves you with work to be done, but moves you farther along the path to true object orientation. 

Since they supplied us with the code, we expect that eventually we could get the Converter to produce code we'd be willing to maintain at a client site years down the road, but with 9000 pretty abstruse lines of code to decipher in the Converter, we generally feel it is easier to start with a clean slate and develop new forms, leaving the @ ... SAYs and GETs behind in 2.x. Besides, hasn't the client always been complaining about one piece of functionality in this screen? And things that need to get moved in that one? And don't you really need to re-engineer the way that query gets built? See what we mean? Start from scratch and give your clients their money's worth. 

After several years of watching people struggle to get their 2.x code to run in VFP, we feel even more strongly about not doing it. However, we also recognize that some people have humongous applications that need to move forward for one reason or another. Our best advice is to check out the incremental conversion methodology developed by MicroEndeavors, Inc. that lets you keep your app running while you slowly move it to VFP. 

You can mix and match the @ commands with Visual FoxPro's controls, so a form could contain both a Visual FoxPro text box and an @...GET. But we can't think of any reason you'd actually want to do this. If you want to put stuff right on the form, use the form's drawing methods: Box, Circle, Line and Print. 

The entries for the individual @ commands do not go into great detail because we don't think you should use them in Visual FoxPro. If you need more information, pick up one of the excellent references available for FoxPro 2.x—see "Resource File" for some suggestions. 

Abs(), Sign()

These two functions let you take the sign off a number. 

	Usage
	nAbsoluteValue = ABS( nValue )

nSign = SIGN( nValue )


ABS() returns the absolute value of the number, that is, the number without the sign. SIGN() does the reverse and returns the sign without the number—it uses 1, 0 and -1 to represent positive, zero and negative signs, respectively. ABS() and SIGN() work on all four number types: currency, double, integer and numeric. 

	Example
	? ABS(37)     && returns 37

? ABS(-37)    && returns 37

? SIGN(37)    && returns 1

? SIGN(-37)   && returns -1

? SIGN(0)     && returns 0


ABS() and SIGN() both handle null values, returning .NULL. 

	See Also
	Int()


_Alignment, _Indent, _LMargin, _RMargin, _Tabs

These variables are vestiges of the printer control system introduced in FoxPro 1.0. Like so much in FoxPro, it seemed like a good idea at the time, but has been superseded several times since. 

These variables control aspects of individual lines produced by ? and ??. Except for _TABS, they're only effective when _WRAP is .T. 

	Usage
	_ALIGNMENT = "LEFT" | "CENTER" | "RIGHT"


No, this variable doesn't determine the political affiliation of the end user. It indicates how text output with ? or ?? should be lined up. Best of all, it works as advertised. 

	Usage
	_INDENT = nCharsToIndent


	[image: image13]
	This one, on the other hand, is flaky. It works, sort of, when applied to memo fields, but not the rest of the time. We say "sort of" because it moves all lines of the memo out to the specified indent, not just the first line. Consider it a "block indent" rather than a "first line indent" and it makes some sense. This, no doubt, ties in with FoxPro's "automatic indent" capability which is so handy when writing code, but it keeps happening even when automatic indent is turned off.


	Usage
	_LMARGIN = nLeftMargin

_RMARGIN = nRightMargin


These variables let you set left and right margins for the print area. Both are measured from the left edge, in character columns of the current output font. 

	Usage
	_TABS = cListOfTabStops


cListOfTabStops is a comma-delimited list of columns. Any CHR(9) in the text being output moves you to the next tab stop. 

All in all, this is a nice set of pretty functional variables. It is remarkable that they work correctly with both proportional and non-proportional fonts. The only problem is they work with ? and ??. We don't use those much anymore, except to sound the bell, where margins and the like aren't particularly relevant. 

	Example
	_LMARGIN = 5

_RMARGIN = 75

_WRAP = .T.

_TABS = "15,25,35"

? "Look"+CHR(9)+"I can"+CHR(9)+"Make"+CHR(9)+"Columns"

? "This is the beginning of a long line that will eventually "+;

 "wrap, showing how the margins work. In fact, showing THAT "+;

 "the margins work."


	See Also
	?, ??, _Wrap


ALines()

This function, added in VFP 6, lets you take character or memo data and dump it into an array, one line per array element. It's faster and easier than using MLINE() and _MLINE for the task. Though each of the two approaches has advantages and disadvantages, the additional functionality added in VFP 7 confirms ALINES() as our first choice for many parsing tasks. 

	Usage
	nLineCount = ALINES( DestArray, cString [, lTrimIt ] 

                     [ cParseChar1, … cParseCharn ] )


	Parameter
	Value
	Meaning

	DestArray
	Array Name
	The array to contain the character or memo data.

	cString
	Memo or Character 
	The string to be broken into lines. Can be a character or memo field or any character expression.

	lTrimIt 
	.T.
	Remove leading and trailing blanks from each line.

	
	.F. or Omitted
	Leave leading and trailing blanks on lines.

	ParseCharx
	Character
	One or more characters to be treated as line breaks.

	nLineCount
	Numeric
	The number of lines in cString, which is the same as the number of rows created in DestArray.



There are all kinds of times when we want to take a multi-line string and break it into its constituent lines for processing. We've had the ability to do so using the MLINE() function and its helper variable, _MLINE, for many versions of FoxPro. But it's always required a bunch of code—and putting the results into an array, which we often want to do, calls for even more code. Enter ALINES(), one function call to do the whole job. 

ALINES() breaks up strings based on explicit line-break characters (either CHR(13) or CHR(10) or a combination of the two) and, starting in VFP 7, on any other characters we specify as well. It doesn't handle breaking long lines into reasonable lengths as MLINE() does, and doesn't pay any attention to the MEMOWIDTH setting. This is both a good thing and a bad thing. When the string is VFP code that you want to execute one line at a time or text from another application that you're processing, not adding line breaks based on MEMOWIDTH is great. When the string is a message that you're trying to make printable, it's a pain. We wish they'd given us an optional line-length parameter (though it would undoubtedly slow the function down). For this reason, if you're trying to break up lines into human-readable lengths (or lengths to fit into character fields), a loop with MLINE() may be the better choice. 

There's one other significant difference between ALINES() and MLINE(). MLINE() strips trailing blanks from each line it creates. By default, ALINES() doesn't. If you need those blanks, this is a big deal. 

	[image: image14]
	Our favorite use for ALINES() is parsing lists of things. For example: 

cColors = "Red, Orange, Yellow, Green, Blue, Purple"

nItems = ALINES( aColors, cColors, .T., ",")

That code requires VFP 7 or later, but you can do the same thing in VFP 6 by including STRTRAN() in the mix:

cColors = "Red, Orange, Yellow, Green, Blue, Purple"

nItems = ALINES( aColors, STRTRAN(cColors, ",", CHR(13)), .T.)


The last two parameters to ALINES() are a little unusual. Normally, you have to include parameters in the order listed, but VFP is smart enough to figure out that a character string as the third parameter means that you're omitting the lTrimIt parameter. 

	Example
	* Assume mField is a memo field.

= ALINES(aAllLines, mField, .T.)

 

* Or use this with a string.

nLines = ALINES(aAllLines, "Here is a string composed of " + ;

                "several lines."+ ;

                CHR(13)+"Here's the second line." + ;

                CHR(10)+"Here's the third line."+ ;

                CHR(13)+CHR(10)+"Notice that it doesn't "+ ;

                "matter which line break character you use.")


	See Also
	GetWordCount(), GetWordNum(), MemLines(), MLine(), _MLine, Set MemoWidth, StrExtract(), StrTran()


_ASCIICols, _ASCIIRows

These system variables control the layout when you use the ASCII clause of REPORT FORM. _ASCIICOLS has logical, intelligent behavior, while _ASCIIROWS, by behaving the same way, is incredibly silly. 

	Usage
	_ASCIICOLS = nColumns

nCurrentColumns = _ASCIICOLS

_ASCIIROWS = nRows

nCurrentRows = _ASCIIROWS


The defaults for _ASCIICOLS and _ASCIIROWS are 80 and 63, respectively, the appropriate size for a U.S. standard portrait page. As you change _ASCIICOLS, the width of the columns in the report and the amount of space between them changes. Be forewarned that some data may be cut off as the number of columns decreases. 

	[image: image15]
	Foolishly, in VFP 6 and earlier, _ASCIIROWS acts pretty much like _ASCIICOLS. As _ASCIIROWS increases, more space is left between rows. As _ASCIIROWS decreases, the rows are put closer together. We think this is incredibly stupid. If you increase _ASCIIROWS, it's because you want more data on the page, not more space between the data.

	In VFP 7, _ASCIIROWS is even stranger than before. You no longer get white space between rows, regardless of the setting. However, the number of rows on the page is fairly random. Things seem to work correctly until _ASCIIROWS is larger than 50. After that, it gets weird. We suspect that all they did was take out the extra rows of white space, but not let you have more rows per page.

	There was one additional bug in the original release of VFP 7 that put the page footer immediately after the detail band. That's fixed in SP1.



Doug comments that he doesn't like the way _ASCIICOLS behaves, either. Having extra spaces between columns is a pain when you want to process the text file afterward. He finds that he needs to set _ASCIICOLS to the exact number of columns in the report to get what he needs. 

Prior to VFP 7, Microsoft didn't see it our way and maintained that the original behavior was correct. Their view was that _ASCIIROWS controls the amount of space taken up by a single row of the detail band. If you make _ASCIIROWS bigger, therefore, it means you want that row to take up more space. Because a row in an ASCII file takes a fixed amount of space, they resolve this by adding blank lines between rows and between repetitions of the detail band. 

We can't imagine why anyone would want this behavior, and are glad Microsoft seems to have come to its senses. However, since the new bug renders _ASCIIROWS totally useless, we're not sure what they're thinking now. 

	Example
	_ASCIICOLS=105  && Landscape width

_ASCIIROWS=48   && Landscape height

REPORT FORM MyReport TO FILE Landscap.txt ASCII


	See Also
	Report


ASelObj()

ASELOBJ() tells you which objects are selected in the Form or Class Designer. With these references, you can determine and/or change an object's properties. In addition, ASELOBJ() allows you to reach "through" an object to determine the object's container or the containing form's data environment. 

	Usage
	nSelCount = ASelObj( ArrayName [, nContainer ] )


	Parameter
	Value
	Meaning

	ArrayName
	Name
	The array to contain the object references.

	nContainer
	Omitted
	Get a reference to each selected object.

	
	1
	Get a reference to the containing object of the selected objects.

	
	2
	Get a reference to the data environment of the form.

	nSelCount
	Positive number
	The number of elements in the array, which is the number of selected objects (or 1 for a container or data environment).

	
	0
	No objects were selected, or in the case of nContainer = 2, the object(s) selected is not on a form. The array is not created in this case.



ASELOBJ() is a key to writing Builders. It lets you determine what control or controls are selected when the user runs a Builder. The object references in the array let you modify the properties and methods of the selected objects based on user input. 

SYS(1270) is a first cousin to ASELOBJ(). It returns a reference to the object currently under the mouse and doesn't require that the object be selected. 

	Example
	* Before typing the code below in the Command Window,

* open the Form Designer and place a few labels on the form.

* Select at least three labels.

= ASelObj(aObjects)

aObjects[1].Caption = "First Label"

aObjects[2].Caption = "Second Label"

aObjects[3].BackColor = RGB(255,0,255)

aObjects[3].ForeColor = RGB(0,255,0)


In a Builder, you can make this sort of change in code based on user input. With references to the form and data environment, you can make changes to those as well—for example, resizing the form to fit the controls it contains. 

	See Also
	Create Form, Sys(1270)


Assist, _Assist

At first glance, ASSIST is a totally useless command; it appears to do nothing. In fact, ASSIST and _ASSIST are a set of very useful hooks that can give you quick access to whatever you'd like. 

	Usage
	ASSIST

_ASSIST = cFileName


ASSIST and _ASSIST were added in FoxPro 2.6 for dBASE compatibility. They provided access to the Catalog Manager. Well, CatMan's gone in Visual FoxPro. The enhanced Project Manager includes the functionality of both the old Project Manager and Catalog Manager. 

So what good are ASSIST and _ASSIST? Like many other system variables, _ASSIST lets you specify a program to run under particular circumstances—in this case, when the ASSIST command executes. The key is that it can be any program at all, and unlike other such variables (say, _GENMENU), there's no built-in time when _ASSIST is executed. It only runs when you tell it by issuing ASSIST. 

For instance, we have several files (tables and text) we need to open each time we start to work on this book. But we're also using VFP for testing as we go. So, it's really handy to have a program that opens all the files we need and sets them up the way we want. By hooking this program to _ASSIST, resetting things is as simple as issuing ASSIST. Using a system variable means the setting doesn't go away when we issue CLEAR ALL to clean up from some disaster. 

	Example
	_ASSIST = "HackSet.PRG"

ASSIST  && runs the program


BackColor, ForeColor

These properties control the color of objects. Big surprise. BackColor is the background (or "paper") color, while ForeColor is the foreground (or "ink") color. For some objects, one or the other is irrelevant. For example, CommandButtons don't have a BackColor, while CommandGroups don't have a ForeColor. 

	Usage
	oObject.BackColor = nColor

nColor = oObject.BackColor

oObject.ForeColor = nColor

nColor = oObject.ForeColor


nColor is a color number in the range 0 to 16777215. Since we don't have all 16 million colors at our fingertips, we generally use the predefined colors in FoxPro.H, RGB() (if we know the right red-green-blue trio) or GETCOLOR() to set these properties. 

Changes to BackColor and ForeColor take place right away. However, text and graphics that have been drawn on a form (either through traditional Xbase commands like ?, DISPLAY, and so forth, or through the form's graphic methods like Line and Circle) don't change color when the form's ForeColor changes. You have to redraw them to change their color. 

If a control's BackStyle is Transparent, its BackColor is ignored and the form's BackColor or an underlying object shows instead. 

	[image: image18]
	Well, almost. In VFP 5 and later, when a text box or edit box with BackStyle set to Transparent gets focus, the control's BackColor shows anyway. 


	[image: image19]
	Be aware that the actual colors you see are affected by factors like video card and resolution. For example, under some circumstances, with certain choices for a form's (or _SCREEN's) BackColor, we see some weird effects with text drawn to the form. Rather than being transparent, there's a box of another color around the text. To see if this is an issue for you, try setting your display to 256 colors, set _SCREEN.BackColor to Magenta (8388863) and then issue DISPLAY MEMORY. We see purple boxes around the DISP MEMO information.



While these properties let you control form and object colors, we strongly advise you not to do so. If you just leave well enough alone, your forms will adopt the user's chosen Windows colors (the ones picked in the Display Properties applet). Almost all the time, that's your best choice. 

	Example
	* You could let the user set a form's colors by putting

* a couple of buttons on the form. The Click code for the

* "Background Color" button might be:

ThisForm.BackColor = GETCOLOR()

* The "Foreground Color" button would have similar code.

* Note that we don't really recommend this approach.


	See Also
	BackStyle, ColorScheme, ColorSource, DisabledBackColor, DisabledForeColor, FillColor, FillStyle, GetColor(), #Include, RGB(), SelectedBackColor, SelectedForeColor


_Beautify, _Browser, _Builder, _Converter, _FoxDoc, _FoxGraph, _Gallery, _GenGraph, _GenHTML, _GenMenu, _GenScrn, _GenXTab, _GetExpr, _ObjectBrowser, _SCCText, _SpellChk, _Transport, _Wizard

These system variables are all hooks. They let you specify a program or application that should run under specific circumstances. A number of them are unused in Visual FoxPro. 

The table below shows the purpose of each hook and its default value. You can change the value of any of these in several ways: Assign it a new value in a program or the Command Window, assign it a value in your Config.FPW file, or, for several of them, set it in the Tools | Options dialog. 

	Variable
	Default value
	Purpose

	_BEAUTIFY
	HOME() + "BEAUTIFY.APP"
	Points to the Documenting Wizard, which is used to produce formatted code and other documentation, like a cross-reference. Unused in VFP 3. 

	_BROWSER
	HOME() + "BROWSER.APP"
	Called when you choose Class Browser from the Tools menu or click the Class Browser button on the Standard toolbar. Don't confuse this variable with the _OBROWSER variable created by running the Browser. That one isn't a system variable and can be released. In VFP 3, this is available only in the Professional edition.

	_BUILDER
	HOME() + "BUILDER.APP"
	Called when you run a builder from any location.

	_CONVERTER
	HOME() + "CONVERT.APP"
	Called when you open a FoxPro 2.x (or older) project, form or report. In VFP 5 and later, also called when you open a VFP 3 project.

	_FOXDOC
	""
	In FoxPro 2.x, called when you choose FoxDoc from the Program menu. Unused in VFP.

	_FOXGRAPH
	""
	In FoxPro/DOS, provided a hook to a graphing package. Unused in VFP.

	_GALLERY
	HOME() + "GALLERY.APP"
	Added in VFP 6. Called when you choose Component Gallery from the Tools menu.

	_GENGRAPH
	HOME() + "WIZARDS\WZGRAPH.APP"
	Points to the Graph Wizard, which provides an interface to MS Graph.

	_GENHTML
	HOME() + "GENHTML.PRG"
	Added in VFP 6. Called when you choose Save As HTML from the File menu.

	_GENMENU
	HOME() + "GENMENU.FXP" or HOME() + "GENMENU.PRG"
	Called when you choose Menu | Generate from inside the Menu Designer and when you build a project including a menu.

	_GENSCRN
	""
	In FoxPro 2.x, called to convert SCX screen files into SPR screen programs. Unused in VFP.

	_GENXTAB
	HOME() + "VFPXTAB.PRG"
	Called by programs created in the Query Designer with the CrossTab check box checked.

	_GETEXPR
	""
	This variable may be one of the coolest additions in VFP 6. It lets you specify your own substitute for the Expression Builder. Whatever you specify is called by GETEXPR.

	_OBJECTBROWSER
	HOME() + "OBJECTBROWSER.APP"
	Added in VFP 7. Called when you choose Object Browser from the Tools menu or click the Object Browser button on the Standard toolbar.

	_SCCTEXT
	HOME() + "SCCTEXT.PRG"
	Added in VFP 5, called to convert forms, classes and other non-text files to a textual format for storage in a source control provider.

	_SPELLCHK
	HOME() + "SPELLCHK.APP" in VFP 6 and earlier. "" in VFP 7. 
	In VFP 6 and earlier, called when you choose Spelling from the Tools menu. In VFP 7, that option is gone. You can set it to call your own spell checker app, if you have one. (Consider writing Automation code to call on Word's Spell Checker.)

	_TRANSPORT
	""
	In FoxPro 2.x, called when you open a project, form, report or label on a platform other than the one on which it was last edited. Unused in VFP—it uses _CONVERTER, above, instead.

	_WIZARD
	HOME() + "WIZARD.APP"
	Called when you choose any wizard from any location.



Note that some of these variables (like _BEAUTIFY) just give you a reference to the program in use—changing these variables doesn't affect what happens when you call the appropriate tool from the menu. (There's one exception here. If you set _BEAUTIFY to the empty string, the menu option is disabled.) Others specify the program that's actually called—changing one of these means that, when you call that tool from the menu, the newly specified program is executed, not the default. In the table, the ones that simply provide a reference use the term "points to," while the others say "called." 

In addition to the automatic ways in which the hooked programs are called, you can run them yourself using DO (_variable). In many cases, you'll need to pass appropriate parameters. See "Builders and Wizards (and Bears, Oh My!)" in Section 5 for the necessary parameters in those cases. Read the source code for VFPXTab and SCCTEXT, and the help file for BROWSER.APP. 

You may also want to consider overriding the default setting of these variables with your own, improved code. 

The unused variables provide a place to store permanent information. Unlike variables you create, nothing you can do releases these variables. They're as persistent as they come. You might use them to store vital information. (If you take this approach in an application, be sure to document it really well. We use these variables primarily to enhance our development environments.) You might also use them as hooks into things you're interested in. See _ASSIST for an example. 

	Example
	* Run a crosstab

DO (_GENXTAB) WITH "Result"


	See Also
	_Assist, _CodeSense, _Coverage, _GenPD, GetExpr, _Startup, _TaskList


Define Box, _Box

DEFINE BOX is one of those commands you should forget you ever noticed. What it's supposed to do is draw a box around output data. In FoxPro/DOS, this command worked more or less as intended. 

	[image: image20]
	As far as we can tell, it doesn't work at all for printed output in Visual FoxPro (nor did it work in FoxPro/Windows); and it works badly when you SET PRINT TO a file.



_BOX determines whether DEFINE BOX actually draws a box or not. For output to file, it appears to work as advertised. 

Don't use DEFINE BOX! Use the Report Designer to create output and take advantage of its box-drawing tool. Use the form's built-in Box() method to draw simple boxes on forms, or the new Shape control if finer control is needed, but avoid this mess at all costs! 

	See Also
	?, ??, @...Box, @...To


Box, Circle, Line, Cls

These form methods let you draw rectangles, circles and ellipses, and lines on forms, and then get rid of them. 

	Usage
	frmForm.Box( [ nLeft, nTop, ] nRight, nBottom )

frmForm.Circle( nRadius [, nCenterX, nCenterY 

                [, nAspectRatio ] ] )

frmForm.Line( [ nLeft, nTop, ] nRight, nBottom )


The position and radius are measured in the form's ScaleMode. If you omit nLeft and nTop for a Box or Line, the current position of the cursor indicated by the CurrentX and CurrentY properties is used. This makes sense, though we think it's kind of odd to have the first two parameters be optional. Similarly, if the center point of the circle is omitted, CurrentX and CurrentY are used. 

nAspectRatio determines whether Circle draws a circle or an ellipse. The aspect ratio is the ratio of the vertical "radius" to the horizontal "radius." A value of 1 gives a circle. Larger values of nAspectRatio lead to an ellipse that is taller than it is wide. Smaller values of nAspectRatio give an ellipse wider than it is tall. 

	[image: image21]
	Circle does weird things with negative aspect ratios. For absolute values up to 1, negative and positive aspect ratios give the same results. But, with an absolute value greater than 1, a negative aspect ratio results in an ellipse larger than one with the same positive aspect ratio. Frankly, the whole thing smells like a bug. Circle shouldn't even accept a negative aspect ratio.



These items are drawn in the form's ForeColor and respect the settings of DrawMode, DrawStyle, DrawWidth, FillColor and FillStyle. 

Shapes created in this way are different than those created with the Shape control. These are just images on the form, like the @ … SAY commands of years gone by. They don't have properties, events or methods. In addition, while the shapes are initially drawn on top of any controls, they can fall behind other objects when the display is refreshed or that control gets focus. 

	Example
	ThisForm.Box(50, 50, 100, 100)

ThisForm.Circle(25, 100, 100)

ThisForm.Circle(40, 20, 70, .5)

ThisForm.Line(100, 200)


	Usage
	frmForm.CLS()


The CLS (for "CLear Screen") method clears away the stuff you draw with Box, Circle, Line, Print and PSet. It doesn't affect actual controls. 

Don't confuse CLS with either the CLEAR command (which visually removes everything from a window, controls and all—the controls come back if you land on them) or the Clear method, which is used with list boxes and combo boxes. 

	See Also
	@...Box, @...Say, @...To, Clear, Clear Method, CurrentX, CurrentY, DrawMode, DrawStyle, DrawWidth, FillColor, FillStyle, ForeColor, Print, PSet, ScaleMode, Shape


Buffering

One of the loudest religious arguments in FoxPro 2.x and older versions was whether to edit fields directly or edit memory variables and then commit the changes to the fields. The proponents for each side could produce all sorts of arguments for why their approach was The Right Way. We're glad to report that the argument is over, and neither side wins. Or maybe both sides win. We definitely win. 

Built-in buffering is one of many cool features of Visual FoxPro. Now, you can write code that appears to edit fields directly, but really works on copies of the fields. When you turn buffering on (either through form properties or with CursorSetProp()), FoxPro maintains several buffers containing the data in its different states. No more SCATTER MEMVAR or GATHER MEMVAR. 

Your code looks like it addresses the fields, but in fact it's really talking to one of these buffers. When the user makes up his mind whether to save or discard his changes, you can either commit the changes to the real table or throw away the buffered changes. One function call (either TableUpdate() or TableRevert()) does the trick in either case. 

FoxPro not only gives you a copy of the data to work on, but it keeps a copy of the original data and the current status of the data, in case someone else changes it while you're working. No more making an extra copy to compare with the network. 

Visual FoxPro has two buffering modes, each of which can be used in two different ways. The two modes are optimistic and pessimistic, referring to the locking scheme used. With the pessimistic approach, a record is locked as soon as you make any change. That way, no one else can change it until you release the lock by committing or reverting the buffer. With optimistic locking, the record isn't locked until you attempt to commit the changes. You run the risk that someone else will make changes at the same time, but records are kept out of circulation for the shortest possible time. There's no need for a religious war over buffering modes because each has its place. Use pessimistic locking for sensitive changes that must go through. Use optimistic locking for everything else. (Views always use optimistic buffering.) 

The other buffering choice is whether to buffer individual records or entire tables. Again, this isn't all-or-nothing. You can use row buffering for some tables and table buffering for others. If you don't specify otherwise, Visual FoxPro uses table buffering when you use a grid for a table, and row buffering for all other tables. 

With a buffered table, you can examine the current value of a field, the value it had when you started working with it (OldVal()—actually, the last time you committed it) and the value it has now on the network (CurVal()). You can also get information (GetFldState()) about the status of any field: Have you changed it? Have you deleted the record it belongs to? Is this a new record? And so forth. You can also find all the records that have changed using GetNextModified(). 

Using the various functions that control all this, you can write code that makes intelligent choices when conflicts arise, and only bothers the user if it doesn't know what to do. 

In addition to buffering, Visual FoxPro lets you wrap updates (both local and remote) in transactions. With a transaction, everything you do is tentative. If one part of an update can't be completed, you can roll the whole thing back. No more worrying about adding the invoice header without the details. Wrap it in a transaction, and you save header and details or nothing at all. 

	See Also
	Begin Transaction, BufferMode, BufferModeOverride, CursorGetProp(), CursorSetProp(), CurVal(), End Transaction, GetFldState(), GetNextModified(), OldVal(), Rollback, SetFldState(), TableRevert(), TableUpdate()


Buttons, ButtonCount

Let me take you a button-hole lower.

William Shakespeare, Love's Labour's Lost 
These properties tell you about the buttons in a button group, whether they're option buttons or command buttons. ButtonCount tells you how many are in the group, while Buttons gives you access to the PEMs of the individual buttons. 

	Usage
	oObject.ButtonCount = nNumberOfButtons

nNumberOfButtons = oObject.ButtonCount

oButton = oObject.Buttons( nButtonNumber )


You can change the number of buttons in a group by changing ButtonCount. If you lower ButtonCount, any extra buttons disappear into oblivion. 

The Buttons collection lets you get at the individual buttons within a button group without worrying about their names. You can look up or change properties or invoke buttons' methods by accessing the button through Buttons. 

To change properties of all the buttons in a group, use the button group's SetAll method instead. 

	Example
	* Look for the button whose caption is "My Favorite Button"

* Assume we're in a method of the button group

LOCAL nButton, lFoundIt

nButton = 1

lFoundIt = .F.

DO WHILE nButton <= This.ButtonCount AND NOT lFoundIt

   IF This.Buttons[nButton].Caption = "My Favorite Button"

      lFoundIt = .T.

   ELSE

      nButton = nButton + 1

   ENDIF

ENDDO

IF lFoundIt

   WAIT WINDOW "My Favorite Button is "+LTRIM(STR(nButton))

ELSE

   WAIT WINDOW "My Favorite Button is missing"

ENDIF


	See Also
	ControlCount, Controls, SetAll


Calculator, _CalcMem, _CalcValue

One of FoxPro's built-in tools plus two system memory variables that track the numbers stored in the calculator—the first stored in the calculator's memory and the second displayed. 

The calculator is a desk-accessory-like utility included with Visual FoxPro, similar to the Calendar/Diary. Microsoft is discouraging its use by removing it from the visual interface of Visual FoxPro and making no effort to improve its awkward appearance, interface or features. The calculator can be invoked by issuing the command ACTIVATE WINDOW CALCULATOR. You can make it available on an as-needed basis by adding the menu pad _MST_CALCU to your custom menu. 

The calculator can be operated completely from the keyboard. For those trivia buffs in the audience, here are the keystrokes that perform the equivalent button functions when the calculator has the focus: 

	Keystroke
	Button Label
	Effect

	A
	M+
	Adds the current value to that stored in memory.

	C
	C
	Press once to erase (clear) the current value displayed, twice to erase the calculation in process.

	N
	±
	Reverses the sign of the current value.

	Q
	Ö
	Returns the square root of the current calculator value.

	R
	MR
	Recalls the value stored in memory.

	Z
	MC
	Clears (zeroes) the value in memory.

	S
	M-
	Subtracts the current value from that in memory.



The calculator takes advantage of a resource file, if one is in use, to store its most recent position. A "Properties" menu option is also available under the standard Edit menu that allows you to store preferences for automatically toggling the NumLock and determining the precision in which numbers are displayed. 

Unless Microsoft begins to show more support for these rather cool little features, we think it might be best for you to consider alternatives like ActiveX controls, or coding your own, to ensure that you have control of the object's behavior now and in future versions. 

	Usage
	nValue = _CALCMEM

nValue = _CALCVALUE

_CALCMEM = nValue

_CALCVALUE = nValue


_CALCMEM stores the value of the calculator memory (what's been stored there), while _CALCVALUE is the current value displayed in the calculator window. 

	[image: image22]
	If you assign values to _CALCVALUE and _CALCMEM, they appear to take. When you activate the Calculator, however, you may find the values displayed as rounded integers—no decimal places are displayed. You can read a value with decimal places just fine from the Calculator, but you can't place one there, unless there was already a decimal place displayed before. This bug appears to be in all versions of Visual FoxPro.

_calcvalue = 123.456

? _calcvalue  && 123.46, rounded to your decimal setting

ACTIVATE WINDOW calculator && You may see "123"

If you clear the Calculator and enter ".0" as a value, and repeat the above lines of code, you should see 123.46 (or more decimals, depending on your setting). What's really interesting is that once you do this, the "setting" is stored in the FoxUser table, so that once it's stored in the FoxUser table, decimals are then always displayed. If you're going to use this feature in your apps, you might want to consider shipping a FoxUser table with your app. You'll need to "fix" your FoxUser first, then copy the record (ID = 'CALCULATOR') into the user's FoxUser table, because the Data field stores its settings in a binary format.


	Example
	SET DECIMALS TO 4

STORE PI() TO _CALCMEM

_CALCVALUE = 2.123

ACTIVATE WINDOW CALCULATOR  

* If you've "fixed" your FoxUser table, you'll see "2.1230" 

* displayed, and pressing MR displays "3.1416".

* If you haven't "fixed" your FoxUser table, you'll see "2" 

* and pressing MR displays "3".


	See Also
	Calendar/Diary, Desk Accessories, Filer, Puzzle, Set Decimals


Calendar/Diary, _DiaryDate

A Mac-like desktop accessory kept in Visual FoxPro for "backward compatibility," but hidden within the product, ignored. 

	Usage
	ACTIVATE WINDOW CALENDAR

DEFINE BAR _MST_DIARY OF _MSYSTEM 

               PROMPT "CALENDAR/D\<IARY"


Having only one application with the ability to do a slew of stuff was handy in the 640K, 20MB hard drive days. Now, with the ability to have many programs running at once in the Windows environment and multi-gigabyte drives considered standard equipment, the idea of a built-in calendar applet seems a little dated. Outlook, Microsoft's personal information manager in the Office package, far exceeds the capabilities of this "mini-app." 

But this application has some neat features, and we are disappointed that Microsoft has chosen to ignore Calendar. The Calendar/Diary option has disappeared from the menu, and you need to dredge through the help file for the example code for the _DiaryDate variable to see how it can be called up programmatically. The fact that a system menu bar is defined for this function appears to be completely undocumented. 

[image: image23]
Figure 1: The Calendar / Diary interface in Visual FoxPro 3.0 and...

[image: image24]
Figure 2: The same interface in Visual FoxPro 7: Bolding of characters is gone.

The interface for using the Calendar / Diary is pretty straightforward. The calendar is displayed on the left-hand side, and buttons across the bottom can be used to change the display to a specific month and year. Note also that there is a "Diary" menu that appears when the Calendar is activated, displaying keyboard shortcuts for many of these functions. In versions before VFP 6, a date with a diary entry associated with it is shown in a bolder font (see figure 1); the month and year are displayed in bold when the calendar is active, and in plain text when the diary is active. In VFP 6 and later, all bolding effects are gone (see figure 2). Diary entries are stored in the currently active resource file. If no resource file is available, a dialog box appears when the calendar is activated for the first time, informing the user he/she will be unable to save diary entries. The entry is stored with a TYPE of "Data," an ID of "DIARYDATA," a NAME of the date in DTOS() format ("YYYYMMDD"), the actual entry in the memo field DATA, and a SYS(2007)-generated checksum in the CHKVAL field. 

Since Microsoft hasn't updated the really ugly interface of this item, we can predict it will probably continue to receive minimal support, perhaps even disappearing in a future release. Using the calendar can be problematic in an application. Storage of diary data in the resource file can cause problems. Because the calendar is a built-in system window, it does not have all of the characteristics or programmability that we may want to integrate into our own application windows, causing difficulties including it within our event handlers. While it may be a handy utility for developers to use, we suggest you consider other sources for supplying a calendar to your users. Several intrepid users have developed all-FoxPro-coded versions of the calendars, available through many of the FoxPro magazines and online services. In addition, there are a several fairly sophisticated ActiveX calendar controls available. 

	Usage
	dDiaryDate = _DIARYDATE

_DIARYDATE = dDateToBeSet


_DIARYDATE is a system memory variable that stores the last date selected in the Calendar/Diary, or the present date if the Calendar has not been used. 

	Example
	STORE {04/11/2000} to _DIARYDATE

? _DIARYDATE


	See Also
	Calculator, Desk Accessories, DToS(), Filer, Puzzle, Set Resource, Sys(2007)


_ClipText

This is a very cool system variable. It contains the contents of the clipboard. What makes it cool is that you can give it a new value, and doing so puts that value on the clipboard. 

	Usage
	_CLIPTEXT = cExpr

cExpr = _CLIPTEXT


In Visual FoxPro for Windows (and FoxPro 2.x for Windows), FoxPro shares the Windows text clipboard. So, not only can you stuff strings into the clipboard to paste into your FoxPro app, you can stuff the clipboard in FoxPro and access that information in other Windows apps. You can also put information in the clipboard in other applications (like CharMap, to get those weird symbol characters) and access it through _CLIPTEXT in FoxPro applications. 

Far more sophisticated use of the clipboard to pass data and more complex structures can be accomplished using the functions exposed in FoxTools or by using an Automation server's DataToClip() method. See those sections for details. 

	[image: image25]
	In versions prior to VFP 7, you can't put memo data directly into _CLIPTEXT, but apply any character function to it and you're in business. TRIM() usually does the trick. 


	Example
	* Stuff the clipboard with the date.

_CLIPTEXT = DTOC(DATE())

* Now you can insert the date by pressing CTRL+V.


	See Also
	DataToClip, FoxTools


CloneObject

This cool method lets you create an exact (well, almost exact) duplicate of an object. The only difference between the two is the name you assign the new one. The only bad thing we have to say about CloneObject is that it's available only at design-time—we can think of all kinds of cool things we could do with it at runtime. 

	Usage
	oObject.CloneObject( cNewName )


The newly cloned object becomes a member of the same container the original object is in. That is, if you clone a label on a form, the new label is added to the same form. 

All the properties and methods of the original are copied to the clone. 

	Example
	* Assume the Form Designer is open and a single object (that

* sits right on the form) is selected.

= ASELOBJ(aObject)    && Get a reference to the selected object

= ASELOBJ(oContainer, 1)     && Get a reference to the form

aObject[1].CloneObject( "NewOne" )

* Now we can reference it through the container

? oContainer[1].NewOne.Name  && Returns "NewOne"


	See Also
	AddObject, AddProperty, ASelObj()


_CodeSense, _FoxCode

These system variables, added in VFP 7, specify the locations of the IntelliSense Manager application and the IntelliSense table, respectively. 

	Usage
	_CODESENSE = cIntelliSenseManager

cIntelliSenseManager = _CODESENSE

_FOXCODE = cIntelliSenseTable

cIntelliSenseTable = _FOXCODE


	Parameter
	Value
	Meaning

	cIntelliSenseManager
	Character
	The filename, including path, of the IntelliSense Manager application. By default, HOME() + "FoxCode.APP".

	cIntelliSenseTable
	Character
	The filename, including path, of the table containing IntelliSense data. By default, "FoxCode.DBF" in the Visual FoxPro subdirectory of the user's profile directory. (In Windows 2000, "C:\Documents and Settings\<user name>\Application Data\Microsoft\Visual FoxPro".)



IntelliSense is surely the most noticeable change in VFP 7. It provides all kinds of ways to make the process of creating code easier. As is true of so much in VFP, IntelliSense is table-driven. The table referenced by the _FOXCODE system variable contains much of the data that makes IntelliSense work. You can change the table contents to make IntelliSense do things your way. 

To make it easier to do so, VFP 7 includes the IntelliSense Manager (accessed through Tools | IntelliSense Manager), an application written in Visual FoxPro. The source code is provided (in HOME()+"Tools\xSource\xSource.ZIP") and, if you wish, you can replace this application with one of your own. Just change the value of _CODESENSE to point to your application. 

You can run the current IntelliSense Manager application by issuing: 

DO (_CODESENSE)

To replace the FoxCode table with one of your own devising, you need to create the new table first, and it must contain at least one record—a version record. The best way to do this is to copy the current FoxCode table, and then get rid of what you don't want. 

Setting _FOXCODE to the empty string turns off IntelliSense entirely. 

See "IntelliSense and Sensibility" in Section 5 for much more on IntelliSense. 

	Example
	_CODESENSE = "MyIntelliSenseManager.APP"

_FOXCODE = HOME() + "MyFoxCode.DBF"


	See Also
	_Beautify, EditorOptions


Row(), Col(), PRow(), PCol()

Back in FoxBase days, even early FoxPro days, we used these functions a lot. We can't remember the last time we used them, though. They return the cursor position on the screen and on the printer, respectively. 

	Usage
	nCurScreenRow = ROW()

nCurScreenCol = COL()

nCurPrintRow = PROW()

nCurPrintCol = PCOL()


You can substitute $ for either ROW() or COL() in @ .. GET and @ .. SAY commands to specify the current location. ROW() and COL() do not respect the settings of ScaleMode, always reporting the row and column coordinates in foxels. 

	Example
	@ROW(), COL()+1 SAY "Here's some information"


	See Also
	@...Get, @...Say, CurrentX, CurrentY


Copy File, Delete File, Erase, Rename

These commands let you do file maintenance without leaving FoxPro. They all do what their names suggest. Delete File and Erase do the same thing—get rid of an existing file. 

	Usage
	COPY FILE SourceFile TO DestFile

DELETE FILE [ FileName | ? ] [ RECYCLE ]

ERASE [ FileName | ? ] [ RECYCLE ]

RENAME OldFileName TO NewFileName


As with any potentially destructive command, be careful when you use these, especially across drives or directories. Starting in VFP 5, DELETE FILE and ERASE have the potential to be very destructive, in fact, since they accept DOS wildcards and let you delete multiple files at once. In addition, they ignore the setting of SET SAFETY—once you issue the command, the file is history. Invoking these commands with the "?" or blank parameter brings up a standard "Select file" dialog. Watch out for this in applications, because the latest incarnations of that dialog give users an awful lot of power via the right-click menu. 

COPY FILE respects SET SAFETY. RENAME ignores SET SAFETY, but it won't let you give a file the name of an existing file. Neither of these commands accepts the "?" to let the user choose a file. However, COPY FILE doesn't scream if you specify ? for the DestFile; it also doesn't do anything. 

DELETE FILE and ERASE don't care whether the file you specify exists. If you give them a nonexistent filename, they just do nothing and go merrily on their way. This is a trap waiting to catch you because it means a typo in a program might keep a file from being deleted with you none the wiser. (Actually, with TALK ON, you do get some feedback to the status bar, but that doesn't help much in a program.) 

Help warns you about renaming tables contained in a database with RENAME. Heed this warning—you'll only create trouble for yourself if you don't. This is what RENAME TABLE is for. 

	Example
	* Copy a file to make a backup

COPY README.TXT TO README.BAK

 

* Change the current transaction file into

* a history file.

cHistFile = "XACT"+TRANSFORM(YEAR(DATE()),"9999")

RENAME XActions.DBF TO (cHistFile+".DBF")

 

* Clean up a temp file. Either of the following will do

DELETE FILE TEMP.TXT

ERASE TEMP.TXT


	See Also
	ADir(), Copy To, Delete, File(), Filer, Rename Table, Set Safety


Set Coverage, Set("Coverage"), _Coverage

These cool features let you keep track of what code has been executed. 

	Usage
	SET COVERAGE TO [ cFileName [ ADDITIVE ] ]

cFileName = SET( "COVERAGE" )

_COVERAGE = cCoverageApp 

cCoverageApp = _COVERAGE

DO (_COVERAGE) [ WITH cLogFile [, lHide ] [, cAddIn ] ]


	Parameter
	Value
	Meaning

	cFileName
	Character
	The name of a file to hold the coverage log.

	
	Omitted
	Turn off coverage logging.

	cCoverageApp
	Character
	The program file to use to analyze a coverage log.

	cLogFile 
	Character
	The name of the file containing the coverage information to process. (The cFileName from SET COVERAGE.) 

	lHide
	.F. or omitted
	The Coverage Analyzer is displayed. 

	
	.T.
	The Coverage Analyzer is hidden.

	cAddIn
	Character
	The path and file name of an add-in utility.

	
	Omitted
	No add-in is specified.



Code coverage was introduced as a partially completed feature in VFP 5.0. While the code coverage logging worked most of the time, Microsoft documented but failed to ship a code coverage analyzer. In addition, there were problems with incorrectly reported form and class methods in version 5.0. Coverage became fully functional in VFP 6. 

The idea behind code coverage is to document every line of code that gets executed in a routine to ensure that all cases have been tested as part of your testing routine. The first step is logging. Issue SET COVERAGE TO YourFileName to begin the process. Each line of code that executes adds a line to this ASCII text file, recording the file name, line, procedure, class and amount of time to execute. When your routine has finished, issue SET COVERAGE TO with no argument to stop recording and close the file. If you want to add to the same file, use the ADDITIVE keyword. 

Now that you have this humongous text file, what to do with it? Run the code coverage analyzer, available on the Tools menu, to read the file and analyze the results. 

The Code Coverage Analyzer is actually two pieces—an engine and an application. The engine does the heavy lifting of parsing the log file, while the application presents the user interface. Realize that you can use the engine, but develop your own interface, to draw out the statistics you need. You could write your own components to do this, or you could consider writing an add-in. Your add-in can be a program, form, APP, EXE, menu or query. You pass the name of your add-in to the Analyzer, and it is invoked after the Analyzer has been instantiated. An object reference to the Analyzer tool is passed to your add-in. At this point, your routine can do whatever it needs to do—change the file to be parsed, add controls to the analyzer form, or present your own form instead. Fortunately, Microsoft has included well-commented source code for all parts of the Coverage application in the Tools\XSource directory and documented the PEMs of the Engine object in Help, so writing add-ins shouldn't be too difficult. 

	Example
	SET COVERAGE TO MYCOVER.LOG

DO MYCODE

SET COVERAGE TO

DO (_COVERAGE) WITH "MYCOVER.LOG"


CPConvert(), CPCurrent(), CPDBF(), IDXCollate(), Set Collate, Set("Collate"), Set NoCPTrans

This slew of functions allows Visual FoxPro to manipulate variables, tables and indexes in a multiple-code-page environment. 

	Usage
	cResult = CPConvert( cFromCP, cToCP, cText )


	Parameter
	Value
	Meaning

	cFromCP
	Integer
	Original code page from which cText is to be translated.

	cToCP
	Integer
	Destination code page for result.

	cText
	Character or memo
	Text to be translated.

	cResult
	Character
	Text after translation.



CPConvert() is not often needed. It accesses the underlying code page translation engine in Visual FoxPro to allow translation of individual phrases. Use this when you have code page-dependent information entered into a table with the incorrect code page for the table. 

	Example
	WAIT WINDOW CPCONVERT(437,1252,"Jos"+CHR(130))


	Usage
	nCodePage = CPCurrent( nWhichOne )


	Parameter
	Value
	Meaning

	nWhichOne
	Omitted or 1
	In Windows, the Windows code page; in other FoxPro platforms, the operating system's code page.

	
	2
	In Windows, the underlying (MS-DOS) code page. In the other platforms, this should return the same as 1.

	nCodePage
	Numeric
	Number of the code page requested.



This function would be far simpler were it not for the idiosyncratic way in which IBM and Microsoft implemented their systems. MS-DOS was here first, and it uses the ASCII code set to produce symbols like the line and box-drawing characters. Windows, on the other hand, chose to base its character sets on the far more widely accepted ANSI characters. Since Windows is merely a thin veneer on top of MS-DOS, programmers on that platform need to be able to detect both code pages and deal with them. 

	Example
	? CPCURRENT(2)  && what's the MS-DOS code page?


	Usage
	nCodePage = CPDBF( [ cAlias | nWorkArea ] )


	Parameter
	Value
	Meaning

	cAlias
	Character
	Alias of the table whose code page is to be returned.

	
	Omitted
	If nWorkArea is also omitted, use the table in the current work area.

	nWorkArea
	Integer
	Work area number of the table whose code page is to be returned.

	
	Omitted
	If cAlias is also omitted, use the table in the current work area.

	nCodePage
	Numeric
	Number of the code page requested.



Use this function to detect the current code page assigned to a table. To change or zero out the code page, use the CPZERO.PRG, located in VFP\TOOLS\CPZERO. To accomplish the same thing in your program, use the GETCP() dialog to determine the correct code page, and then call the CPZERO.PRG, which you can include within your application, to change the code page. 

	Example
	* The Wizard table should be flagged for US Windows codepage

DO CPZERO WITH HOME()+"\Wizards\Wizard.DBF",1252

USE Wizard.DBF IN 0 ALIAS DaWiz

? CPDBF("DaWiz")  && should return 1252


	Usage
	cCollation = IDXCollate( [cCDXName, ] nIndexNumber

               [, cAlias | nWorkArea ] )


	Parameter
	Value
	Meaning

	cCDXName
	Character
	The name of the compound index file to examine. If no file is specified, IDXCollate() works on currently open indexes.

	nIndexNumber
	Integer
	As is typical of index functions, this refers to a number assigned to each open stand-alone index, in the order they were opened, then the tags of the structural index, then any other compound indexes open, with each tag enumerated in the order created.

	cAlias
	Character
	Alias of the table whose index collation is to be returned.

	
	Omitted
	If nWorkArea is also omitted, use the table in the current work area.

	nWorkArea
	Integer
	Work area number of the table whose index collation is to be returned.

	
	Omitted
	If cAlias is also omitted, use the table in the current work area.

	cCollation
	Empty string
	The number of the index specified exceeds the number of open indexes, or the work area number specified has no open file.

	
	Character 
	The collation sequence which applies to this index.



A collation sequence is the set of rules that determine the order in which values within the index will be presented. The most familiar (and the default) is MACHINE, which is a straight binary ordering of the values stored within the index. See Set Collate below for more details on this. 

	Example
	USE HOME() + "\Wizards\BuilderD.DBF"

? IDXCOLLATE(1)  && returns "MACHINE"


	Usage
	SET COLLATE TO cSequence

cSequence = SET( "COLLATE" )


	Parameter
	Value
	Meaning

	cSequence
	Character
	Specifies the sequence in which items should be sorted.



We Americans do not appreciate some of the conveniences of computing using the English language. Unfortunately, in being English-centric, many of the early developers of computers made data interchange between different languages a real challenge. Sort order is one very good example of this. In both Germanic and many Romance languages, the sort order is dictated not just by the individual value or weight of each character, but also the values of its nearby characters or the addition of diacritical marks or ligatures to those characters. A straight sort by the binary value of individual characters, even when correctly translated by code page into the correct alphabetic sequence, won't do it. 

A caution: Recently we've run into a few cases where the performance of FoxPro's queries is far slower than we'd expect. In some cases, joins between tables where one table is lacking the proper index results in missing records. In all of these cases, the problem was that the indexes were created with a non-Machine collation. Our thumbrule: Always create index tags with the Machine sequence if they are to be used in Select statements. 

Each collation sequence has specific ordering rules, needed for a specific purpose. For example, to list an alphabetical sequence of names in German, SET COLLATE TO "GERMAN". A handy ordering sequence for case-insensitive sorts is the GENERAL sorting order. 

	[image: image26]
	All collation sequences except Machine use 2 bytes per key so the maximum size of a key expression is only 120 bytes, not 240 as Help states.


	Example
	SET COLLATE TO "GENERAL"


	Usage
	SET NOCPTRANS to cFieldList

cFieldList = SET( "NOCPTRANS" )


SET NOCPTRANS is a relic from the 2.x days, where a field that should be treated as containing binary data, and not translated from one code page to another, was flagged by explicitly issuing the SET NOCPTRANS command each and every time it was open. Needless to say, somewhere in their code, nearly everyone forgot to issue the command, and binary data got translated into garbage left and right. Visual FoxPro solves the entire problem by allowing data to be specified as "Binary" within the Table Designer and with the NOCPTRANS keywords of the CREATE TABLE and ALTER TABLE commands. 

	See Also
	GetCP(), Index, Set CPCompile, Set CPDialog


ObjNum(), ObjVar(), _CurObj

Here's another batch of stuff that was crucial in FoxPro 2.x and is obsolete in Visual FoxPro. In FoxPro 2.x, each item (@ ... GET or @ ... EDIT) on a screen was assigned a unique "object number." Within button groups, each button received its own number. These two functions and one system memory variable operated on those object numbers. 

OBJNUM() takes a variable listed in @ ... GET or @ ... EDIT and returns its object number. OBJVAR() (added in version 2.6) takes an object number and returns the name of the variable associated with it. _CUROBJ controls the focus—it contains the number of the object with focus. Changing _CUROBJ changes the focus. 

Visual FoxPro has alternative ways to accomplish these tasks, and the object-number system is only partly there. Visual FoxPro forms don't use it. Forms that include any @ ... GETs or @ ... EDITs do, but OBJNUM() and OBJVAR() don't recognize any Visual FoxPro controls on the form, only the GETs and EDITs. 

The bottom line is: don't mix and match. Use Visual FoxPro's new form methodology. Refer to objects on the screen by their object hierarchy (Form.Page frame.Page.Object) rather than a number, and move focus to the desired object by calling its SetFocus method. If you can't do this for some reason, then stick completely with the old @-based system. 

	See Also
	@ Commands, ActiveControl, SetFocus, This


CurrentControl, DynamicCurrentControl, Sparse

These properties (together with the actual objects in the column) determine what control you see in any particular grid cell. If a column has more than one control specified, CurrentControl indicates which should appear, unless DynamicCurrentControl has a value. In that case, the expression specified for DynamicCurrentControl determines which control shows up. 

Sparse specifies whether the current control (whichever property determines it) appears all the time in every row for that column or only when that cell has focus. 

	Usage
	grcColumn.CurrentControl = cControl

cControl = grcColumn.CurrentControl

grcColumn.DynamicCurrentControl = cExpr

cExpr = grcColumn.DynamicCurrentControl


A grid column can contain as many controls as you like to allow different modes of displaying or entering the same information. These two properties determine which of the multitudes the user actually sees. 

Like the other Dynamic... properties, DynamicCurrentControl expects a character string containing an expression. The expression itself should evaluate to the name of a control as a character string. (This is yet another place where we're glad FoxPro offers multiple sets of string delimiters.) 

If DynamicCurrentControl evaluates to a control the column actually contains, that value supersedes any value in CurrentControl. Although DynamicCurrentControl is pretty dynamic, fortunately it's not so dynamic that the control actually changes while you're sitting on it, even if the expression's value changes. 

You can't use "This" in the expression to refer to the grid column. That's because the expression gets evaluated not in a method of the column, but at the grid or form level. You need to use ThisForm.TheGrid'sName.TheColumn, instead. 

	Example
	* Say a particular column has a ControlSource of nFld. You can

* put a text box and a spinner in the column and show the user

* the text box for odd values and the spinner for even values.

* Why? Why not? 

ThisForm.grdMyGrid.grcNFld.DynamicCurrentControl= ;

   "IIF(MOD(nfld,2)=0,'spnNFld','txtNFld')"

 

* If, on the other hand, we'd just like to have the user see the

* spinner every time:

ThisForm.grdMyGrid.grcNFld.CurrentControl = "spnNFld"


	Usage
	grcColumn.Sparse = lHideControl

lHideControl = grcColumn.Sparse


One of the coolest features of Visual FoxPro is putting any control you want in a grid. However, looking at a grid full of combo boxes and spinners and check boxes could drive someone nuts. That's where Sparse comes in. When you set Sparse to .T. (the default), the current control for a column shows only when a cell of that column has focus and shows only for the cell with focus. In many cases, it's a lot nicer to look at only the value as text until you're ready to edit it. Then, you want to see the combo or check box or whatever. 

	When CurrentControl for a cell is a check box bound to a logical field and Sparse is .T., nothing at all shows when the cell doesn't have focus. It's empty. Since check boxes aren't too ugly, your best bet is probably to set Sparse .F. for those columns.



The control that shows when Sparse is .T. and a cell doesn't have focus is a text box. But you don't have any control over it at all. It's not the default text box that gets put in the column when you create the grid. It's a different text box, internal to FoxPro. 

Why do you care? Because, even if the control you're using for a column is a text box, when Sparse is .T., things you specify for that text box apply only to the current cell. By "things," we mean properties like InputMask and Format that you might want to see all the time. Fortunately, in VFP 5 and later, columns have Format and InputMask properties that apply when Sparse is .T. and the cell doesn't have focus. Columns also have their own BackColor and ForeColor that apply in the same circumstances. These colors also propagate down to the contained controls, unless those controls have their own. 

If you need more control than that (or you're in VFP 3), forget about Sparse. Instead, put both the control you want and a text box in the column. Format the text box the way you want it and set DynamicCurrentControl to an expression that chooses the text box when the cell doesn't have focus and the other control when it does. You need to compare ActiveColumn for the grid to the particular column's ColumnOrder and check ActiveRow against the record number (if you're working with records in natural order). Our example below shows how to do this. 

	Example
	* In the same grid as above, here's the DynamicCurrentControl

* expression to use the spinner when the cell has focus and the

* text box the rest of the time:

IIF(ThisForm.grdMyGrid.ActiveColumn = ;

    ThisForm.grdMyGrid.grcNFld.ColumnOrder AND ;

    ThisForm.grdMyGrid.ActiveRow = RECNO(),"spnNFld","txtNFld")

* And, of course, we'll need:

ThisForm.grdMyGrid.grcNFld.Sparse = .F.


	See Also
	Column, DynamicAlignment, DynamicBackColor, DynamicFontBold, DynamicFontItalic, DynamicFontName, DynamicFontOutline, DynamicFontShadow, DynamicFontSize, DynamicFontStrikeThru, DynamicFontUnderline, DynamicForeColor, DynamicInputMask, Grid


_DblClick, _IncSeek

These system variables control two aspects of the user interface. _DBLCLICK determines how long the user can pause between mouse clicks and still have it considered a double-click. _INCSEEK determines the speed for incremental search in list and combo boxes. 

	Usage
	_DBLCLICK = nSeconds

_INCSEEK = nSeconds


nSeconds can range from .05 (1/20th of a second—we don't know anybody who can double-click that fast) to 5.5 (handy for someone with a physical disability that makes double-clicking or typing quickly difficult). Because the ability to double-click varies widely from one person to the next, you'll make your users really happy if you let them set _DBLCLICK through your application. 

Of course, we and our users would be even happier if this setting were controlled by the similar setting in the Windows Control Panel. Happily, as of VFP 7, it is: _DBLCLICK gets its initial value from the setting chosen by the user in the Mouse applet. 

Prior to VFP 7, _DBLCLICK had a second use: controlling the speed of typing for incremental search. We're delighted that Microsoft decided to follow our advice and create a new system variable for this item. _INCSEEK (which also draws its initial value from the double-click setting in the Mouse applet!) determines how much time can pass between keystrokes and still take advantage of the incremental search feature of combos and lists. 

Since users' ability to type may be quite different from their clicking ability, you may want to uncouple _INCSEEK from _DBLCLICK and provide a facility for the user to adjust this value in your applications. 

	Example
	_DBLCLICK = 1   && 1-second delay

_INCSEEK = ThisForm.spnIncSeek.Value


	See Also
	ComboBox, DblClick, IncrementalSearch, ListBox, Set Mouse


Soundex(), Difference()

These two functions are both for comparing character strings and they're both pretty useless. By now, they're just being dragged along into new versions for backward compatibility. 

Both DIFFERENCE() and SOUNDEX() are intended to help find matches when strings might be misspelled. SOUNDEX() applies a coding scheme also called Soundex to a string and returns a four-character string consisting of a letter and three digits. The Soundex encoding is meant to give matching results for strings that are phonetically identical or nearly so. So, for example, SOUNDEX("Smith") and SOUNDEX("Schmidt") both return "S530." However, the scheme isn't really good enough to be useful in practice. Among other things, the letter returned is always the first letter of the original string, so "Knowles" and "Noles" return two different values. 

DIFFERENCE() takes two strings and returns a number between 0 and 4, measuring the phonetic difference between the strings. A value of 0 means the strings are very different, while 4 means they're quite similar. Unfortunately, like SOUNDEX(), DIFFERENCE() isn't very smart. It also doesn't know about things like "kn" or that "ph" sounds like "f." We've never found a reason to use DIFFERENCE(). 

If you really need to match strings that might be misspelled, take a look at a terrific library for Visual FoxPro called PhDbase by Korenthal Associates. It handles "fuzzy" search, as well as quick searches of full text. That's what we use when we need those abilities. 

	Usage
	cReturnValue=SOUNDEX(cString)

nReturnValue=DIFFERENCE(cString1, cString2)


	Example
	? SOUNDEX("Tamar")   && returns "T560"

? SOUNDEX("ted")     && returns "T300"

? DIFFERENCE("tamar","ted")   && returns 2

? DIFFERENCE("ted","teddy")   && returns 4


	See Also
	Like()


_Dos, _Mac, _Unix, _Windows

These system variables tell you which platform FoxPro is running on. In Visual FoxPro for Windows, _WINDOWS is true and the rest are false. In VFP/Mac, _MAC is .T. If we ever see Visual FoxPro for the other platforms, we expect we'll see the matching system variables set true, but we're not holding our breath. 

	Usage
	lIsDos = _DOS

lIsMac = _MAC

lIsUnix = _UNIX

lIsWindows = _WINDOWS


Back when cross-platform meant more than Windows 95 and Windows NT, these variables were useful for cross-platform development. They allowed you to bracket platform-specific code so it didn't cause errors on the other platforms. 

	Example
	DO CASE

CASE _WINDOWS

   MODIFY WINDOW screen FONT "Arial",10

CASE _MAC

   HIDE WINDOW screen

ENDCASE


	See Also
	OS()


DoScroll

DoScroll is a method to scroll a grid horizontally and vertically as the user would with the scrollbars. It was introduced with grids in VFP 3, so we were really surprised this method was not extended to forms when forms gained scrollbars in VFP 6. 

	Usage
	grdGrid.DoScroll( nDirection )


	Parameter
	Value
	Meaning

	nDirection
	0
	Simulates a click on the scrollbar up arrow.

	
	1
	Simulates a click on the scrollbar down arrow.

	
	2
	Simulates a page-up click—clicking in the space above the thumb and below the scroll arrow.

	
	3
	Simulates a page-down click—clicking in the space below the thumb and above the scroll arrow.

	
	4
	Simulates a click on the left scrollbar arrow.

	
	5
	Simulates a click on the right scrollbar arrow.

	
	6
	Simulates a form-left click—clicking in the space to the left of the thumb and the right of the scroll arrow.

	
	7
	Simulates a form-right click—clicking in the space to the right of the thumb and the left of the scroll arrow.

	
	Omitted or any other number
	Ignored.



A useful alternative to including scrollbars on the grid itself, DoScroll can be used to cause the same effect without all the visual interference, or with an alternative user interface. The Scrolled event fires after DoScroll is called, and receives a parameter for direction matching that is set by this method. 

	Example
	grdGrid.DoScroll(0)


	See Also
	Grid, Scrollbars, Scrolled


Low-Level File Functions

Low-level file functions (LLFFs) are cool. They can be a hacker's best friends, and are certainly essential tools in our toolbox. With LLFFs, you can open and read all those nasty binary files that the operating system and manufacturers would prefer you didn't take apart. With LLFFs, it's possible to translate a Windows Cardfile into a .DBF, read and rewrite a FoxPro macro file, translate the Screen memory variable stored in a .MEM file, or study and perhaps rearrange the FoxPro codepage sort logic (FOXPRO.INT). 

See the disk for the result of a few hours of taking apart the .FKY keyboard macro function file and analyzing the results: PeekFKY.PRG dumps the contents of a macro file to the screen. Check out the topics SAVE MACROS and RESTORE MACROS, and think about what you could do with keyboard macros being updated programmatically. 

Low-level file functions follow a logic that will be familiar to users of other languages, but may be a little different for FoxPro devotees. Rather than referring to a file by its alias, a file "handle" number is created using FOPEN() or FCREATE(). A file pointer, like a record number, determines where the next bytes are read or written to. The FSEEK() function allows us to move this file, repositioning the pointer relative to the current position or to the top or bottom of the file. FREAD() or FGETS() reads data from the file, either byte-by-byte or by lines, and FWRITE() or FPUTS() matches the reading functions with their equivalent outputs. Finally, a few miscellaneous functions allow flushing (FFLUSH()), end-of-file testing (FEOF()) and error checking (FERROR()). 

Low-level file functions were the most common ways to manipulate files that were not in DBF format, but there are several other techniques, some new to Visual FoxPro. These techniques may have advantages over LLFFs for your task, and bear consideration. If you are working with ASCII text, consider sticking the file in a memo field with APPEND MEMO and using the memo field functions MLINE() and MEMLINES() to hack it apart, or an APPEND FROM TYPE SDF to pull it into multiple records. If the text files are in the Windows INI format, consider the WinAPI calls described in "INI Files." New to VFP 6 are the StrToFile() and FileToStr() functions to load an entire file into one character variable in one swoop. 

If you'll be writing out a text file, consider SET ALTERNATE and SET TEXTMERGE as well as using LLFFs. 

But for hacking binary formatted data, there's nothing better than the low-level file functions. Let's see what they can do. 

FChSize()

FCHSIZE() modifies the size of a file on disk. This command can increase, decrease or zero-out the size of a file. 

	Usage
	nNewSize = FCHSIZE( nFileHandle, nSize )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	nSize
	Numeric
	Final desired size.

	nNewSize
	–1
	An error has occurred; see FERROR() for more information. Check for disk space, file opened with write access, or network rights issues.

	
	Numeric
	The new size of the file. 


	Example
	lnRetVal = FCHSIZE(8, 0)


The above example zeroes out the length of the file opened with file handle 8. 

FClose()

FClose() closes a file opened with low-level file functions. 

	Usage
	lSuccess = FCLOSE( nFileHandle )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	lSuccess
	.F.
	An error has occurred; see FERROR() for more information. Check for the correct file handle, network rights.

	
	.T.
	File successfully closed.



Note that issuing the CLOSE ALL command closes all files opened with the low-level file functions. 

	Example
	* Close the file opened with file handle 8.

llRetVal = FCLOSE(8)


FCreate()

FCREATE() creates a file on disk, and returns a "handle"—a numeric value that other low-level file functions must have to refer to the same file. 

	Usage
	nHandle = FOPEN( cFilename [, nAttribute ] )


	Parameter
	Value
	Meaning

	cFilename
	Character string
	The full path and filename of the file to be opened.

	nAttribute
	0 or omitted
	Read-Write. Any option other than the default 0 parameter prevents FoxPro from writing to the file.

	
	1
	Read-only.

	
	2
	Hidden.

	
	3
	Read-only and hidden.

	
	4
	System.

	
	5
	Read-only and System.

	
	6
	System and Hidden.

	
	7
	Read-only, Hidden and System.

	nHandle
	–1
	An error has occurred; see FERROR() for more information.

	
	Positive integer
	File successfully opened.


	Example
	* Create the AUTOEXEC.BAT files to be read/write. 

lnHandle = FCREATE("C:\AUTOEXEC.BAT", 0)


If you create a file and forget the handle number, such as by issuing the command =FCREATE("test",0), you can determine the file handle of all open files with the DISPLAY STATUS command. 

No warning is given to overwriting an existing file, even if SAFETY is SET ON! 

FEOF()

FEOF() tests for end-of-file on a file opened with low-level file functions. 

	Usage
	lAtEof = FEOF( nFileHandle )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	lAtEof
	.F.
	Not at end-of-file.

	
	.T.
	At end-of-file.


	Example
	* Check for end-of-file for the file opened with file handle 8.

llRetVal = FEOF(8)

	FEOF() returns .T. if passed an invalid file handle. Neither a FoxPro error condition (like error 1113: "no file is opened in that area, stupid") nor a low-level file error (see FERROR(), for example 6: "Invalid file handle") is generated.


FError ()

Low-Level File Functions error-handling is funny but useful. Rather than use FoxPro's built-in error generation and the language's error-handling capability, LLFFs set an error condition detectable by FERROR(). This allows local testing of the success or failure of LLFF execution, where a global error-handling method may not be appropriate. No arguments are accepted. 

	Usage
	nErrNum = FERROR()


	Error Number
	Meaning

	2
	File not found.

	4
	Too many files open (limited by file handles).

	5
	Access denied.

	6
	Invalid file handle given.

	8
	Out of memory.

	25
	Seek error (usually top of file).

	29
	Disk full.

	31
	Error opening file.


	Example
	nFileHand = FOPEN("NUL:")

if nFileHand < 0  && negative means error

do case

  case FERROR() = 2  && File Not Found...


FFlush ()

FFLUSH() flushes data to disk. 

	Usage
	lSuccess = FFLUSH( nFilehandle )


	Parameter
	Value
	Meaning

	nFilehandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	lSuccess
	.T.
	FLUSH successful.

	
	.F.
	Error.


	Example
	llRetVal = FFLUSH(8)


The above example flushes all data to disk for the file opened with file handle 8. 

FGetS()

FGETS() reads data from a low-level file. FGETS reads characters until it reaches the first of these three limits: the number of characters specified in the second parameter, the first (next) occurrence of a carriage return, or the end of the file. 

	Usage
	cString = FGETS( nFileHandle [, nBytes ] )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	nBytes
	Numeric
	Maximum number of bytes to read; defaults to 254 if not specified.

	cString
	Character string
	Characters read.


	Example
	lcRetVal = FGETS(8,400)


The above example reads characters from file handle 8, starting at the current file position, and ending at the first occurrence of a carriage return, the specified number of bytes, or the end-of-file (hint: test FEOF()). 

FOpen ()

FOPEN() opens a file on disk for reading and/or writing, and returns a "handle"—a numeric value that other low-level file functions (LLFFs) must have to refer to the same file. While it may be theoretically possible to open other DOS devices, such as "LPT1:" or "COM1:" with this command, results under Windows are less than 100 percent predictable (and get less predictable under Win98 and NT!). A far better solution to serial port manipulation is to use the Win API function calls (i.e., OpenComm()) or a third-party library specially designed for serial-port work. 

	Usage
	nHandle = FOPEN( cFilename [, nMode ] )


	Parameter
	Value
	Meaning

	cFilename
	Character string
	The full path and filename of the file to be opened.

	nMode
	0 or omitted
	Read-only, buffered.

	
	1
	Write-only, buffered.

	
	2
	Read/write, buffered.

	
	10
	Read-only, unbuffered.

	
	11
	Write-only, unbuffered.

	
	12
	Read/write, unbuffered.

	nHandle
	–1
	An error has occurred; see FERROR() for more information.

	
	Positive integer
	File successfully opened.


	Example
	nHandle = FOPEN("C:\AUTOEXEC.BAT", 0)


The above example opens the AUTOEXEC.BAT files to be read, with buffering turned on. Buffering is normally the preferred method of access. 

FPutS()

FPUTS() writes data to an opened low-level file. FPUTS writes characters until it reaches the smaller of these two limits: the number specified in the third parameter, or the total length of the string supplied. 

	Usage
	nRetVal = FPUTS( nFileHandle, cString [, nBytes ] )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	cString
	Character
	Character string to be written to file.

	nBytes
	Numeric
	Maximum number of bytes to write; defaults to the length of the string supplied plus an ending carriage return-line feed pair.

	nRetVal
	0
	An error occurred. Test with FERROR(), check for correct file handle, file rights.

	
	Integer
	The number of bytes written.


	Example
	* Write characters to file handle 8,

* starting at the current file position,

* ending at the end of the string or 

* the specified number of bytes 

lnRetVal = FPUTS(8,MyTestStr,127)


The return, lnRetVal, reflects the number of characters written, including a final carriage return (CHR(13)) and line feed (CHR(10)). 

FRead()

FREAD() reads data from a low-level file. Unlike FGETS(), FREAD() just reads the number of characters specified, ignoring the value of those characters. (FGETS() is sensitive to carriage returns and line feed characters.) FREAD() stops after reading the number specified in the second parameter, or encountering the end-of-file. 

	Usage
	cString = FREAD( nFileHandle, nBytes )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	nBytes
	Numeric
	Maximum number of bytes to read.

	cString
	Character string
	Characters read.


	Example
	* Read characters from file handle 8,

* starting at the current file position, 

* ending at the specified number of bytes, or the end-of-file 

*(hint: test FEOF())

lcRetVal = FREAD(8,400)


FSeek()

FSEEK() repositions the file pointer within a low-level file. This is sort of the low-level file function equivalent of the GO command for DBFs. 

	Usage
	nPosition = FSEEK( nFileHandle, nLocation [, nRelativeTo ] )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	nLocation
	Numeric
	Location at which to position the file pointer.

	nRelativeTo
	0 or omitted
	Measure nLocation from the beginning of the file.

	
	1
	Measure nLocation from the current file pointer position.

	
	2
	Measure nLocation from the end of the file; that is, start at the end and move nLocation characters toward the beginning.

	nPosition
	Integer
	The new position within the file, expressed as the number of bytes offset from the beginning.


	Example
	* Moves the file pointer to a position 400 characters 

* from the beginning of the file

lcRetVal = FSEEK(8,400)

=FSEEK(8,0,0)  && GO TOP

=FSEEK(8,0,2)  && GO BOTTOM

lnWhereAmI = FSEEK(8,0,1)  && return current file position


FWrite()

FWRITE() writes data to a low-level file. Unlike FPUTS(), FWRITE() writes exactly what you specify, without appending any characters (such as carriage returns or line feeds). Use FWRITE() when you are trying to write data out to a very specific file format, such as Windows CardFile, FOXPRO.INT or another binary format. 

	Usage
	nRetVal = FWRITE( nFileHandle, cString [, nBytes ] )


	Parameter
	Value
	Meaning

	nFileHandle
	Integer
	The file handle created with FOPEN() or FCREATE().

	cString
	Character
	The data to be written to the file.

	numBytes
	Numeric
	Maximum number of bytes to write.

	nRetVal
	0
	An error has occurred. Check for correct file handle, string, disk space and rights.

	
	Integer
	The number of characters actually written.


	Example
	nRetVal = FWRITE(8,REPLICATE("Fred",200),400)


The above example writes 400 characters to file handle 8, starting at the current file position. 

	See Also
	Append Memo, Close All, FileToStr(), MemLines(), MLine(), Set Alternate, Set TextMerge, StrToFile()


MacDesktop

This property doesn't do anything in the Windows version of Visual FoxPro. In the Mac version, it controls whether or not a form is a child of the main Visual FoxPro window. Use ShowWindow to control this in the Windows version. 

	See Also
	Desktop, MDIForm, ShowWindow


Name

This property is one of the keys to OOP. It's what you use to refer to an object so you can look at its properties, change them, or call a method. 

	Usage
	oObject.Name = cName

cName = oObject.Name


Every object in Visual FoxPro has a Name property, except for those created in weird ways like using SCATTER NAME. 

You can assign a name to an object in the Property Sheet when you add it to another class, or in its own Init method (or the container's Init method). Although you can change the Name of an object at runtime, it's usually not a good idea once you get past the Init method. If you have code that depends on an object's Name and you change Name, the code fails. 

If you don't assign an object a Name, FoxPro does it for you. That's why the first text box you add to a form is Text1 and the next is Text2, and so on. With classes defined in code (rather than a VCX), there can be a tremendous performance penalty for letting FoxPro do it in versions prior to VFP 7. See "Faster Than a Speeding Bullet." 

	Despite the warning above, in a code class, don't ever assign Name a value in the properties section unless it's the same as the class name. Instead, assign the new instance a Name in Init. When you do the assignment in the properties section, it changes the name of the class, not the name of instances of the class. 



Although you usually use Name to refer to an object, you don't do that for the outermost object that gets created via either CREATEOBJECT() or DO FORM. For that object, use the variable holding the object reference—then, you can drill down into the object using the names of the various members. 

	Example
	DEFINE CLASS MyTextBox AS TextBox

 

* Don't do this!

* Name = "anything"

* But do this for speed issues, if using a coded class.

Name = "MyTextBox"

 

PROCEDURE Init

 

* Do it this way instead, or the class name changes.

This.Name = "anything" 

 

ENDPROC

 

ENDDEFINE


	See Also
	Caption, Class, CreateObject(), Do Form, Scatter


Object

This keyword lets you dig inside an OLE Container control to talk to the actual ActiveX control inside. Most of the time, though, you don't need it. 

	Usage
	oleContainer.Object.Property = uValue

uValue = oleContainer.Object.Property

oObject = oleContainer.Object

oleContainer.Object.Method()


To put an ActiveX control (formerly known as an OLE control) on a form, you first drop an OLE Container control (one of the VFP base classes) onto the form, then choose the appropriate ActiveX control. If you prefer and you've registered the control with VFP, you can choose the control from the ActiveX controls toolbar, but it still brings an OLE Container with it. Both the OLE Container and the ActiveX control have properties and methods. 

When you're writing code, you can almost always ignore the fact that there are two levels of objects here, and just refer to the ActiveX control's PEMs as if they belong to the OLE Container. However, in a few rare cases, this approach doesn't work—that's when you need the Object keyword. Stick it after the reference to the OLE Container to make sure you're addressing the ActiveX control, not its container. 

The most obvious place that Object is needed is when you're actually trying to get an object reference to the ActiveX control. For example, the ActiveX TreeView control uses an ActiveX ImageList. You have to set the TreeView's ImageList property in code, giving it a reference to the ImageList you've already created. In this case, the Object keyword is required. 

	Example
	* Say you've put an ImageList on the 

* form and called its OLE Container 

* oleImageList. To assign this 

* ImageList to a Treeview in an

* OLE Container called oleTree, use:

ThisForm.oleTree.ImageList = ThisForm.oleImageList.Object


	See Also
	OLEControl


Objects

This collection property gives you a handle to all the objects instantiated in an instance of VFP. It's convenient for writing generic tools and especially useful for Automation servers. It also lets you look inside VFP's container classes. 

	Usage
	oObject = oContainer.Objects[ nIndex | cName ]


For the application-level collection, you can access Objects either through the _VFP system variable that references the current VFP session or through the Application object. (Every control contains a property that points to the VFP instance that contains it.) Objects contains an element for each object that's been created in the VFP session in question. So what's an object in this context? Let's start with the easy stuff. Anything you create with CREATEOBJECT() or NEWOBJECT() counts as an object. So does any form you run with DO FORM. Then, it gets more interesting. Browses count because, behind the scenes, a browse is really a grid. Opening the Class Designer adds a formset object to the collection—we're not sure why it's a formset rather than an object of the class you're working on. Maybe it refers to the designer rather than the object being designed. Opening the Form Designer adds two objects to the collection: a data environment object and a formset object. Finally, opening the Report Designer or Label Designer adds only a data environment object to the collection, because there are no Report or Label objects. 

	You can manipulate at least some of the design-time objects that show up in the _VFP.Objects collection, but in VFP 6, we found it a dangerous game to play. We crashed or froze VFP more often in a couple of hours playing with the Objects collection than in weeks of doing other stuff. So far, in VFP 7, this collection seems a lot more stable.



All of VFP's container classes have their own collections to let you see what's inside. (For example, Form and Column both have Controls collections, while PageFrame has Pages.) However, they all also have an Objects collection, which makes it much easier to write generic code. In VFP 5 and VFP 6, only those containers that can contain objects of more than one type have an Objects collection in addition to their native collection. The objects that are limited to a single type of child (like OptionGroups, which are composed only of OptionButtons, or Grids, which are composed of Columns) have only their native collections. 

Since we could already drill down into those objects with their native collections, why would we use Objects? Primarily for consistency. It keeps us from having to remember the name of the native collection (though that's not a big deal, since it's usually Controls). More importantly, though, Objects is a common collection in the Automation world. By including it, VFP makes things easier for programmers who want to talk to VFP but don't know it that well. 

The Objects collection has a Count property that tells you how many objects are in the collection at the moment. Objects also has an Item method that allows you to iterate through the collection by number or name. This is redundant in VFP, where we can just specify the index directly on the Objects collection, but we suspect it's a standard feature of COM objects like Objects. 

	In earlier versions, getting access to the Objects collection at runtime is flaky. As long as you're drilling down directly from the _VFP application object, all is well. But if you create an object reference to, say, a form, VFP can't see the Objects collection. That is, 

? _VFP.Objects["frmMyForm"].Objects.Count 

tells you the number of controls on the form. But:

oForm = _VFP.Objects["frmMyForm"]

? oForm.Objects.Count

gives an error message. This bug is fixed in VFP 7.

	Unfortunately, there's a nice new bug to replace it. When you use an object reference to drill down, and use the Item method to address the members of the collection, you get an error unless you got the object reference by drilling down from _VFP. That is, with a form running, this code works:

? _VFP.Objects["frmMyForm"].Objects.Item[1].Name

but this doesn't:

oForm = _SCREEN.ActiveForm

? oForm.Objects.Item[1].Name

And, assuming the form is called MyForm, neither does this:

? MyForm.Objects.Item[1].Name

We think this is a deeper manifestation of the earlier bug. Fortunately, you can work around it by using the shorthand notation:

? oForm.Objects[1].Name

	But wait, there's more. IntelliSense mishandles the Objects collection. At the application level (that is, for _VFP.Objects), no IntelliSense is available—that is, no members pop up after you hit the period. Below that level, only Count is offered as a choice (except, sometimes, when Value appears instead). The Item method never shows up, and when you specify an index, IntelliSense doesn't offer you the members of the specified object.


	Example
	* Find out what objects exist.

FOR EACH oObj IN _VFP.Objects

   ? oObj.Name

ENDFOR


	See Also
	Application, Buttons, ButtonCount, Columns, ColumnCount, Controls, ControlCount, Count Property, CreateObject(), Forms, FormCount, NewObject(), Pages, PageCount, _VFP


_PAdvance

This used to be one of the most useful of the printer variables. It let you determine whether page breaks were made up of a single form-feed character (Ctrl+L or CHR(12)) or as many linefeeds (CHR(10)) as needed to get to the top of the next page. This one saved an awful lot of people with odd-sized paper. 

But, as far as we can tell, it doesn't work with the Report Designer (though we're pretty sure it did in FoxPro 2.x). Since that's the way to go for reporting, _PADVANCE has become obsolete. (It does actually work with @SAY reports, but who wants to write those?) 

	Usage
	_PADVANCE = "FORMFEED" | "LINEFEEDS"


	Example
	_PADVANCE = "FORMFEED"

SET DEVICE TO PRINT

SET PRINT TO TEST.TXT

FOR nBatch = 1 TO 20

   FOR nRow = 1 TO 10

      @nRow,1 SAY "Line "+PADL(nRow,2)

   ENDFOR

ENDFOR

SET PRINT TO

SET DEVICE TO SCREEN

* Check Test.Txt and you'll see the CHR(12)'s after each batch


	See Also
	Eject, _PLength


Page, PageFrame

Multi-page tabbed dialogs were one of the hottest things around when VFP was introduced. Now it seems they're everywhere. Visual FoxPro's version is called a PageFrame. It's a container object that contains Pages. Pages are also containers and can contain any controls, even other PageFrames. 

We do think page frames are pretty cool looking. We also think they're terribly overused. Like so many of the other widgets in Windows, page frames are useful in dialogs. The Tools-Options dialog is the prime example of this, but many builders also make good use of page frames (and those guys are all built in Visual FoxPro). However, you can go too far with this. A page frame is heavily mouse-oriented—it has no place in heads-down data entry. (Neither do a lot of the other fancy controls, but that's not what we're talking about here.) 

Use page frames where they make the resulting form clearer. The builders are a good example of this—while everything in the builder is related, each page contains a set of items that are closely related to each other. 

Although tabbed interfaces are all the rage, you can also make tabless page frames. We can imagine using them for some of the things we used to do with multi-screen sets where all the screens occupied the same location. 

PageFrame

	Property
	Value
	Purpose

	ActivePage
	Numeric
	The number of the page currently on top. This number is based on PageOrder, not creation order.

	Objects
	Collection
	A COM collection containing references to the individual pages in the page frame. Also, has a Count property.

	PageCount
	Numeric
	The number of pages in the page frame.

	PageHeight, PageWidth
	Numeric
	The size of an individual page.

	Pages
	Collection
	References to the individual pages in the page frame.

	Tabs
	Logical
	Determines whether the individual pages have tabs at the top.

	TabStretch
	Numeric
	Determines whether multiple rows of tabs are used when the tabs don't fit in a single row.

	TabStyle
	Numeric
	Determines whether all tabs are the same width or each is sized to accommodate its Caption.


	Event
	Purpose

	Moved
	Fires when the page frame's position is changed programmatically.


Page

	Property
	Value
	Purpose

	ActiveControl
	Object
	Reference to the control on the page that has focus.

	Caption
	Character
	The text on the page's tab. Can include a hotkey.

	ControlCount
	Numeric
	The number of controls on the page.

	Controls
	Collection
	References to the individual controls on the page.

	Objects
	Collection
	A COM collection containing references to the controls on the page. Also has a Count property.

	PageOrder
	Numeric
	The display position of this page in the page frame.

	The documentation for VFP 5 and later says that Page has the KeyPreview property. (Page is listed in the Applies To list of the KeyPreview topic.) They did in VFP 3 (though it didn't do anything there). It was removed in VFP 5.



Page frames and pages refresh differently than other controls. When the Refresh method for a page frame is called, it calls only the Refresh method for the ActivePage. The other pages don't get visually refreshed. 

We can see why the Refresh methods don't all fire up front—it could really bog things down. But why the heck don't they get refreshed when they come to the top? How are your changes supposed to filter through? 

Under the circumstances, we don't see ourselves putting much code in the Refresh method for an individual page. We'll use the page's Activate method or the controls' UIEnable events instead for things we want to be sure happen. We'll also put This.Refresh() in each page's Activate method to make sure the controls get refreshed. (We'd love to handle all this with a nice subclass of PageFrame, but read on.) 

Both Page and PageFrame have the MouseEnter and MouseLeave events added in VFP 7. These fire alternately, not together. That is, when you enter a page frame (from any direction), its MouseEnter event fires. But then, the page frame's MouseLeave fires, and MouseEnter fires for the active page. Same thing as you exit – the page's MouseLeave fires, then the page frame's MouseEnter, then the page frame's MouseLeave. 

There's one additional complication here. The tabs are considered to be part of the page frame, not the page. So if you enter over a tab, the page frame's MouseEnter fires. It's not until you leave the tab area and enter the page that the page frame's MouseLeave and the page's MouseEnter fire. 

Okay, one more complication. When you change pages with the keyboard, MouseLeave for the old active page and MouseEnter for the new one do not automatically fire. It's not until you move the mouse that those events occur. 

Subclassing PageFrame is a real pain. You can create subclasses, but there are several nasty restrictions on them. 

First, if you use the Class Designer, they always contain Pages—you can't subclass Page and stick your subclass in a page frame. Page is one of the "half-classed" objects discussed in "OOP is Not an Accident." In fact, you can create subclasses of Page in code, and even create subclasses of PageFrame that use your Page subclasses, but you can't do it visually. More important, when you add pages to the page frame by upping PageCount, the new ones are based on Page, not your subclass. There's no way to tell a PageFrame subclass to always use a particular Page subclass. This is a serious limitation. However, FoxPro programmers, always willing to go the extra mile, have found a solution. You can put code in the page frame class's Init method to remove the base class pages and substitute your custom pages. You can even set all the properties as they were in the base pages, so that you can do your design work in the Form Designer. Going this way, though, you lose the ability to add custom code to the individual pages, since there's no way to add code to an object at runtime. 

Second, when you put your subclassed page frame in a form in the Form Designer, you can't reduce the number of pages—the only way to get rid of a page is to RemoveObject at runtime. 

No big deal, you say. You'll just use a subclass with no pages and change PageCount in the Form Designer. But then you lose most of the benefits of subclassing in the first place—you can't define your default page and start out with it. 

Next problem. If you leave the pages on the page frame, when you drop the class onto a form or create an instance, you can't change the names of the individual pages. 

Fundamentally, there's no good way to create a single subclass of PageFrame and use it wherever page frames are called for. Your best choices are to put a lot of code in your PageFrame subclass to modify pages as they're instantiated or to subclass Page in code and use AddObject at runtime to add your subclass to the page frame. We're not thrilled with either choice, but it's looking more and more like Microsoft isn't going to get around to making Page a first-class citizen. 

	Early versions of VFP had some problems dealing with changes to a page's Name after you've started messing with it. All the code to set page properties is stored in the Properties memo of the page frame that contains it. In earlier versions, when you set some properties for the page, and then changed the page's name, some of the property assignments used the original default name (Page1 or whatever) while some used the new name.


	Example
	* All the changes shown here would be made

* in the Property Sheet.

* Set a PageFrame to have 3 pages.

* With the PageFrame selected:

PageCount = 3

 

* Set a page to have white on blue text

* and to have a tab of "Sky".

* With the page selected:

BackColor = RGB(0,0,255)

ForeColor = RGB(255,255,255)

Caption = "Sky"


	See Also
	ActiveControl, ActivePage, Caption, ControlCount, Controls, KeyPreview, MouseEnter, MouseLeave, Moved, Objects, PageCount, PageHeight, PageOrder, Pages, PageWidth, Tabs, TabStretch, TabStyle, UIEnable


_PageNo

This variable is that rare bird—a system variable that works in the most up-to-date way of getting output, but not very well in the older, obsolete ways. It contains the current page number and can be used to put the page number on a report. Although you can use it with ? and ?? output, _PageNo's value there is based solely on the number of lines output and _PLength—it doesn't notice when you issue EJECT (though EJECT PAGE does properly update _PageNo). With @..SAY output, we can't get anything sensible from _PageNo. 

	Usage
	_PAGENO = nPageNumber


An often-asked question is, "How do I put 'Page X of Y' on my report?" The Report Writer doesn't support this natively, but you can use _PageNo to create a workaround. First, set up a variable or property to store the total pages in the report (remember, it must be in scope while the report runs). Initialize it to 0, then put it into your report's footer after the page number, in the "Page X of Y" format. When you run the report, you'll do it twice, first with the NOCONSOLE and TO FILE options. When it finishes the first pass, store the value of _PageNo to the variable or property. Now run it again to its normal destination, and the total number of pages properly displays on the report. 

	Example
	* To implement Page X of Y on a report, you

* might use the following expression:

"Page "+LTRIM(STR(_PAGENO))

 

 

* In the report footer, use an expression like the following:

"Page "+LTRIM(STR(_PAGENO))+" of "+LTRIM(STR(nTotalPages))

* Then run the following code:

PRIVATE nTotalPages REPORT FORM MyReport TO FILE Temp.RPT NOCONSOLE

DELETE Temp.RPT

NTotalPages = _PAGENO

REPORT FORM MyReport TO PRINT


	See Also
	Eject Page, _PLength, Report


_PColNo

This variable indicates the current output column for output created with ? and ??. It's analogous to the PCOL() function used with @...SAY output. 

	Usage
	_PColNo = nStartColumn


You can set _PCOLNO to start ?? output in a specified column. 

	Example
	SET PRINT TO test.txt

SET PRINT ON

 

FOR ncnt = 1 TO 20

   _PCOLNO = 2*ncnt

   ?? "Line "+PADL(ncnt,2)

   ?

ENDFOR

 

SET PRINT OFF

SET PRINT TO

* now examine Test.Txt to see the effect of setting _PCOLNO


	See Also
	?, ??, PCol()


_PLength, _PLineNo

These variables are part of the printer control system for streamed output produced with ? and ??. _PLENGTH determines the length of a page, and _PLINENO determines the current line number. 

None of this has anything to do with reports generated in the Report Designer. 

	Usage
	_PLENGTH = nPageSizeInLines

nPageSizeInLines = _PLENGTH

_PLINENO = nNextLineToUse

nNextLineToUse = _PLINENO


_PLENGTH and _PLINENO interact with the ON PAGE command to determine when that command's event fires. 

Setting _PLINENO doesn't change where the next line appears. It simply changes where the streaming output engine thinks it's printing. 

	Example
	* See ON PAGE


	See Also
	?, ??, Eject Page, _LMargin, On Page


PSet, Point

These two form methods give you fine control over colors. PSet lets you color an individual pixel. Point tells you the color of a specified pixel. 

	Usage
	frmForm.PSet( [ nXCoord, nYCoord ] )

nColor = frmForm.Point( [ nXCoord, nYCoord ] )


	Parameter
	Value
	Meaning

	nXCoord, nYCoord
	Numeric
	The coordinates of the point of interest.

	nColor
	-1
	The specified point is not in the form.

	
	0 - 16,777,215
	The color of the specified point.



If the coordinates are omitted for either method, the values of CurrentX and CurrentY are used. PSet sets the specified point to the current ForeColor. PSet interacts with DrawWidth—if DrawWidth is more than one, multiple pixels get colored. (In fact, at the resolutions we work at, we can't even see the colored pixel when DrawWidth is 1. Either that or no pixel gets colored when DrawWidth is 1.) The docs say the colored pixels are centered on the specified point. That's wrong in VFP 3—in that version, the colored pixels use the specified point as the upper-left corner. 

	[image: image36]
	In VFP 5 and later, when DrawWidth is greater than 2, the point you specify doesn't actually get colored. Instead, as we said, you get a box centered on the specified point. The confusing thing is that applying Point to that point immediately after "painting" it with PSet gives you not the foreground color, but the background color. See the examples to see how this works. In fact, FillStyle and FillColor determine what color that point actually is. It's as if you called the Box method.



These methods are not affected by ScaleMode—they always address individual pixels. 

	Example
	* Color a few points

_SCREEN.ForeColor = RGB(255, 0, 0)

_SCREEN.PSet(100, 100)    && It's red, but hard to see

? _SCREEN.Point()           && Returns 255

_SCREEN.Cls

_SCREEN.DrawWidth = 10

_SCREEN.PSet()            && Now you can see it

? _SCREEN.Point()         && Returns the value of BackColor

? _SCREEN.Point(100, 100) && So does this

? _SCREEN.Point(95, 95)   && This, however, gives you 255

? _SCREEN.Point(95, 104)  && So does this and a number of others

? _SCREEN.Point(-10, -200)   && Returns -1 


	See Also
	Box, CurrentX, CurrentY, DrawWidth, FillColor, FillStyle, ForeColor, ScaleMode


_PPitch, _PQuality

These don't do anything. They may have been somewhat useful, a long time ago, in a galaxy far, far away.

_PSpacing

This system variable determines the line spacing for output generated with ? and ??. 

	Usage
	_PSPACING = nLineSpacing


You can set _PSpacing to anything you want, as long as it's between 1 and 3. Only the integer part of the value is significant—no line-and-a-half spacing allowed. 

	Example
	SET PRINT TO test.txt

SET PRINT ON

 

_PSPACING = 2

FOR ncnt = 1 TO 20

   _PCOLNO = 2*ncnt

   ?? "Line "+PADL(ncnt,2)

   ?

ENDFOR

 

SET PRINT OFF

SET PRINT TO

* Note that the line spacing is doubled


	See Also
	?, ??


Rand()

Ain't life Rand? This function returns random numbers. Actually, they're technically known as "pseudo-random" numbers. This means that you can produce the same sequence over and over. It's important because, when testing, you need the ability to work on the same data. 

	Usage
	nReturnValue = RAND( [nSeed] )


	Parameter
	Value
	Meaning

	nSeed
	Any negative number or zero
	Seeds the function with a value based on the system clock.

	
	Any positive number
	Seeds the function with the passed value.

	
	Omitted
	Seeds the function based on the last value returned by it. If this is the first call, the seed is 10001.



All random-number functions need a seed value based on which the function computes a random result. Passing the same number should (and does, in FoxPro) return the same "random" number. This ability allows you to test on the same data set. 

To get the most random results, pass a negative number or zero as the initial seed. That seeds the function based on the system clock. On subsequent calls, omit the seed and FoxPro uses the previous number as a seed. You'll always get the same sequence if you start with the same value, but with a negative initial seed, that's very unlikely. 

To mathematicians, a "random number" is always between 0 and 1, and that's what RAND() returns. Since you'll usually want a random integer in some range, you need to scale the result appropriately. The formula to convert the result to an integer between 0 and some boundary seems simple: 

INT(nBoundary*RAND())

but things are seldom what they seem. 

In all versions of FoxPro before VFP 5, RAND() returns both 0 and 1, but very rarely. Using the formula above, you get many fewer occurrences of nBoundary from a sequence of calls to RAND() then any other number in the range. You can fix that by adding 1 to the result (INT(nBoundary*RAND())+1), but then you get too few zeroes. There's no good solution to the problem. The best solution is for RAND() to never return one or the other of the endpoints (that is, RAND() should either never return 0 or never return 1). 

Good news. Starting in VFP 5, that's exactly how it's done. RAND() never returns 1, so you can use INT(nBoundary * RAND()) to get values from 0 to nBoundary-1 or INT(nBoundary * RAND()) + 1 to go from 1 to nBoundary. No more ugly correction code that fouls up statistics. 

One question you'd think we'd hear a lot is "How random are the results of RAND()?" There are various tests you can run to test the randomness of a sequence of random numbers. The downloads (www.hentzenwerke.com) contains two tests. Each generates a sequence of random numbers and then computes a statistic about it. The first is the Chi-Squared test. It returns a value you can look up in a table (try a statistics textbook, or check out Excel's CHIDIST() function) to see how random RAND() is. The second test is called the "Coupon Collectors" test. The return value is the average number of random numbers you have to generate in order to be sure you've gotten at least one of each value in a range. 

	Example
	* Here's the right way to use RAND() in a program, if you 

* want a truly random sequence.

* First, seed RAND() with -1 (or 0 or any other negative number)

RAND(-1)

 

* Then, use RAND() with no seed inside whatever loop 

* you're performing.

FOR nCnt = 1 TO nIterations

  nRand = RAND()

  * Now do something with nRand

ENDFOR


Report, Label

These commands let you generate reports and labels created with the Report and Label designers. They have a multitude of options—some fairly new, others as old as Xbase itself. 

	Usage
	REPORT FORM ReportFileName | ?

       [ Scope ] [ FOR lForExpression ] 

       [ WHILE lWhileExpression ]

       [ RANGE nStartPage [ , nEndPage ] ]

       [ PREVIEW [ [ WINDOW DefiningWindow ] 

          [ IN  WINDOW ContainingWindow ]

          | IN SCREEN ]

         [ NOWAIT ]  

         | TO FILE OutFileName [ ASCII ]

         | TO PRINTER [ PROMPT ] ] 

       [ NOCONSOLE ] [ NOEJECT ]

       [ HEADING cHeadingText ] [ PLAIN ] [ SUMMARY ]

       [ ENVIRONMENT ] [ PDSETUP ]

       [ NOOPTIMIZE ]

       [ NAME ObjectName ]

LABEL [ FORM LabelFileName | FORM ? ]

       [ Scope ] [ FOR lForExpression ] 

       [ WHILE lWhileExpression ]

       [ PREVIEW [ IN SCREEN ] [ NOWAIT ] 

         | TO FILE OutFileName [ ASCII ]

         | TO PRINTER [ PROMPT ] ] 

       [ NOCONSOLE ] 

       [ ENVIRONMENT ] [ PDSETUP ] [ SAMPLE ]

       [ NOOPTIMIZE ]

       [ NAME ObjectName ]


The "FORM" in REPORT FORM and LABEL FORM stands for "format" and, in fact, you can still write it that way. More importantly, the term "format" makes it clear what these commands do—they produce a report or a label based on a specified format. 

LABEL has a neat trick. If you just type LABEL without FORM, you're prompted to choose a label file. For both labels and reports, specifying ? does the same thing. 

The Scope, FOR and WHILE clauses limit output to the records meeting those conditions. See "Xbase Xplained" for details. 

Report and label output can be sent various places. By default, it appears in the active window. You can suppress that output with NOCONSOLE. 

Specifying PREVIEW puts the output in the Print Preview window. Preview mode is much better in VFP than in FoxPro 2.x, with a dockable toolbar, an assortment of zoom ratios and a NOWAIT clause that lets you put up a report and then continue a program. Starting in VFP 5, you can also define a window for the Preview and give it a custom title and appearance (much as you can for a Browse). Then, you can use the WINDOW clause to put the preview in the custom window. 

VFP 6 added even more options. First, the IN WINDOW option lets you confine the preview to a specified window. Second, when you have a top-level form, issuing PREVIEW without specifying a window puts the preview in that window—this is a major improvement over previous versions, which put the preview into the main VFP window, often hidden in a top-level-form-based application. Finally, the IN SCREEN clause is for those rare cases where you're working in a top-level window, but want the preview in the main VFP window. We can't quite imagine that situation, but we're sure it'll help someone, somewhere. 

The LABEL command doesn't offer the WINDOW clause, but does preview in a top-level window and supports the IN SCREEN clause, even though it's not documented. 

The biggest remaining weakness in report previews is that you can only zoom up to 100%. We can think of plenty of situations where we'd like to zoom to 150% or 200% or even 400%—we sure do it in other applications. Also, the ability to specify the exact zoom magnification would be nice, rather than being locked into a few choices. Word's Print Preview features options like "Page Width" and "Text Width" that speed the ability to see everything at the highest magnification. 

There are two ways to send report or label output to a file. If you specify TO FILE without the ASCII clause, all the codes for the specified printer go with it. You can then send that file to the printer at your leisure. The ASCII clause, on the other hand, lets you create a file that a human can read, instead of one filled with gobbledygook printer control. The output with ASCII is VFP's best guess of a character-based layout from a graphical one—this option is best for columnar reports without many fonts, sizes or multiple rows of information to line up. Not only that, it's fast—really fast. On the other other hand, the formatting you get with ASCII is pretty basic. 

Finally, you can send your report to the printer. Add the PROMPT clause to let the user choose a printer; otherwise, FoxPro uses the printer you specified in Page Setup or the Windows default, if you didn't change it. There are several problems with the PROMPT clause in VFP 5—they're documented in the Microsoft Knowledge Base, so we won't repeat them here. 

The RANGE clause lets you indicate which pages of a report should be produced. It's ignored when you specify PREVIEW, but does affect output to the screen, printer or a file. Very handy for those times when one page of a 30-page report gets jammed in the printer. In VFP 7, the upper limit for RANGE has been raised to 65,534 (yeah, right, like someone's going to read a report that long). 

NOEJECT and PDSETUP have something in common. They don't do anything with Visual FoxPro reports. Both have their origins in FoxPro for DOS, where they were very handy. 

ENVIRONMENT is almost as useless. It applies only to reports converted from older versions of FoxPro, where you didn't have properties like AutoOpenTables available. With the ENVIRONMENT clause, the report's data environment is opened regardless of the setting of AutoOpenTables. 

HEADING and PLAIN also date way, way back. HEADING was designed to let you specify a page heading for a report when you run it—it let you send things like "First Quarter 1995" and so forth, putting what you send on the last line of the Page Header band. HEADING still works, but it doesn't pay any attention to what's already on that line, clobbering any text already there. We recommend you avoid it and simply use expressions in your reports. PLAIN eliminates page headings from the report—note that you get no page headings at all, not even on the first page, as Help says. 

SUMMARY omits the detail band. All the other bands appear. It's one way to get (surprise) summary information. 

The SAMPLE clause of LABEL doesn't do a thing and, in fact, is no longer documented. In FoxPro/DOS, it let you print a test label to see if the labels were properly aligned in the printer. Microsoft doesn't believe anyone uses dot-matrix printers with Windows, so you can't do this. Instead, we recommend you run a test set in a loop with LABEL FORM MyLabel NEXT nHowManyToASheet until the operator is happy with the results. 

NOOPTIMIZE slows down your report. See SET OPTIMIZE if you think you might need to use this. 

Last, but by no means least, is the NAME clause. Although reports and labels are not objects (in the OOP sense), their data environments are. The NAME clause lets you assign a name to the DataEnvironment of a report or label. You can use that name in method code (which seems like a bad idea to us—use This instead) or from the Command Window or a program; with the NOWAIT clause of PREVIEW, your program can still be running while the preview is displayed. You can also reference the data environment from code running in the entry and exit events for the report's bands. We haven't ever actually used these abilities in an application, but it's nice to know they're there. By the way, if you don't include the NAME clause, you can still reference the data environment by using the report name. 

	Example
	* Put a report preview in a specified window.

DEFINE WINDOW RepPreview FROM 0,0 to 50,100 ;

   title "Employees"

REPORT FORM Employees PREVIEW ;

   WINDOW RepPreview NOWAIT

LABEL FORM Customer TO PRINT

REPORT FORM Employees TO FILE emps.txt ASCII

REPORT FORM SaleHist TO PRINT RANGE 5,7


Since VFP 3, when you printed reports, the Print Spooler dialog displayed "Visual FoxPro Report" while the report was queued to print. Starting in VFP 7, the name of the report or label is passed to the Print Spooler dialog. Now the dialog doesn't give away what language your application was written in, and it also makes it far easier to delete specific reports from the queue when you know their names. 

	See Also
	_ASCIICols, _ASCIIRows, Compile Label, Compile Report, Create Label, Create Report, DataEnvironment, Modify Label, Modify Report, Set Optimize


_RunActiveDoc, Sys(4204)

This system variable and function are related to getting active document applications to run. _RunActiveDoc contains the Active Document launcher program. SYS(4204) controls active document-debugging mode. 

	Usage
	cActiveDocLauncher = _RunActiveDoc

_RunActiveDoc = cActiveDocLauncher

Sys( 4204 [, nOnOrOff ] )


	Parameter
	Value
	Meaning

	cActiveDocLauncher
	Character
	The name of a program to run to launch active document applications. Defaults to "RunActD.PRG" in the VFP home directory.

	nOnOrOff
	0
	Turn off debugging mode for active document applications.

	
	1 or omitted
	Turn on debugging mode for active document applications.



Because active document applications can run in VFP itself or in a browser, testing and debugging them presents some special problems. 

First, the code you need to run an active doc app is different depending on whether you want it to run in a browser, in the VFP runtime, or in the VFP development environment. Rather than force us to figure it all out, the VFP team has provided us with a clever little tool that offers a jumpstart. When you choose Tools | Run Active Document from the menu (in VFP 6 only) or issue DO (_RunActiveDoc), a little dialog appears that lets you choose an active doc application and indicate where it should run. 

If you don't like the application provided, though, you can replace it with one of your own by changing the value of _RunActiveDoc. You can make the change either programmatically or through the File Locations page of the Tools | Options dialog. Press the Set As Default button on that page to save your preference to the Registry, whether you use the dialog to set it or do it programmatically. Otherwise, you'll have to reset it every time you start VFP. 

Take a look at the program Microsoft supplies—there are some interesting routines for reading and writing configuration information to the resource file, as well as a glimpse of how Microsoft localizes its applications. 

So much for letting us test our active doc apps. The second problem is debugging them. If your app is running in a browser, how can you use the debugger to see what's happening? That's what SYS(4204) is about. When you set it to 1, the debugger can see the app running in the browser and lets you check things out. When SYS(4204) is set to 0, the debugger doesn't know a thing about your browser-hosted app. 

	[image: image37]
	Unlike most of the SYS() functions, SYS(4204) doesn't return either the current or the new value. In fact, there's no way to determine the current value of this setting. You'd think, by now, Microsoft would know that for every setting, there should be an equal and opposite way to find the current setting.



Incidentally, the way the active doc launcher provides the Browser (Debugging) mode is by turning on SYS(4204). No magic. 

	Example
	_RunActiveDoc = "MyGreatActiveDocLauncher"

SYS(4204,1)  && Turn on debugger in browser mode.

DO (_RunActiveDoc)


	See Also
	ActiveDoc, Run Method


_Samples

This system variable, added in VFP 6, contains the path to the sample code that comes with Visual FoxPro. 

	Usage
	_SAMPLES = cPathToSamples


In VFP 6, MSDN took over VFP's sample code, and locating it wasn't as simple as looking in HOME() + "\SAMPLES", especially since both VFP and MSDN can be installed wherever you want to put them (and, once installed, MSDN's paths are long and difficult to remember). This variable was added to make it easier to find the samples when you want to look them over for ideas, take advantage of some item located there in your own code, or just use them for giving demos. In VFP 7, the samples again belong to the FoxPro directory tree (a consequence of pulling VFP out of Visual Studio, no doubt), but we still have this handy variable, which is much simpler to type than a path, anyway. 

HOME(2) contains the same value. In fact, changing _SAMPLES changes the return value of HOME(2). Of course, changing _SAMPLES doesn't move the samples; so think twice before you change the variable. The location of the samples is stored in the Registry and _SAMPLES gets its value from there, but changing _SAMPLES, then using Tools | Options and choosing Set as Default does store the new value. 

	Example
	USE (_SAMPLES + "TasTrade\Data\customer" )


	See Also
	Home()


Scatter, Gather

These commands were part of every application we wrote in FoxPro 2.x (and even in earlier versions). We don't use them the same way in Visual FoxPro because VFP's buffering capabilities make it unnecessary, but one VFP-only version of the commands can be useful. 

SCATTER copies all or part of a record to an array, memory variables or an object. GATHER collects data from an array, variables or an object and sticks it in the current record. 

	Usage
	SCATTER [ FIELDS FieldList | 

        [ LIKE Skeleton1 ] [ EXCEPT Skeleton2 ] ]

        MEMVAR | TO ArrayName | NAME ObjectName 

        [ MEMO ] [ BLANK ]

GATHER [ FIELDS FieldList | 

        [ LIKE Skeleton1 ] [ EXCEPT Skeleton2 ] ]

        MEMVAR | FROM ArrayName | NAME ObjectName

        [ MEMO ]


	Parameter
	Value
	Meaning

	FieldList
	List of fields
	The fields to include in the command.

	Skeleton1
	Fieldname with wildcards
	A specification for which fields to include in the command. Wildcards are * and ?.

	Skeleton2
	Fieldname with wildcards
	A specification for which fields to exclude from the command. Wildcards are * and ?.

	ArrayName
	Name
	An array to hold the scattered data or from which to get the gathered data. SCATTER creates or enlarges the array, as needed, but doesn't shrink it.

	ObjectName
	Name
	An object that has a property corresponding to each specified field. SCATTER creates the object.



Both the LIKE and EXCEPT clauses can be included in a single command, but they can't be mixed with an explicit field list. 

There are three forms of these commands that differ in where they put or get the data: MEMVAR, TO array and NAME. MEMVAR uses a set of memory variables with the same names as the fields. SCATTER MEMVAR creates the memory variables; GATHER MEMVAR collects data from them. 

Using TO or FROM indicates that field data is stored in an array. SCATTER creates or redimensions the array, if necessary. GATHER collects data from the array, matching it element by element with the record. 

The NAME clause lets you create an object that has a property for each specified field. SCATTER creates and populates the object. GATHER collects data from the object. Note that objects created this way differ from almost every other object you can create in Visual FoxPro. Unless the table being SCATTERed happens to have such fields, these objects have neither Name nor Class properties. They have no methods or events. Listing memory shows the object as type Object with class "EMPTY." These "lightweight" objects are ideal candidates to use as parameters, having no visible presence and no surprising behaviors. 

The MEMO clause indicates that memo fields should be included in the command. Without it, memos are ignored. When you're working with arrays and omit MEMO, you need to be careful that you don't get mixed up about which column refers to which field, since memo fields are skipped. General fields are always skipped, so watch out for those, too. 

BLANK says to create the memvars, array or object, but populate it with blank values of the appropriate type. In a FoxPro 2.x application, you'd typically SCATTER MEMVAR BLANK at the beginning of an Add operation, then APPEND BLANK and GATHER MEMVAR when the user asked to save. 

Visual FoxPro appears to fix a "by design" bug in older versions. In FoxPro 2.x, if you SCATTER MEMVAR, then call a routine that makes one of the SCATTERed memvars private and GATHER in that routine, if the called routine contains no assignment to the private memvar, FoxPro would find the memvar in the calling routine and GATHER it even though the PRIVATE declaration should have hidden the memvar in the calling routine. Visual FoxPro respects the privacy of the variable. This change should only create a problem for you if you depended on the old behavior. 

We don't use SCATTER and GATHER much in Visual FoxPro. Instead, we use row and table buffering and work with the fields directly. However, we do know people who use the NAME clause to create "record objects" so that they can write more OOP-y code. 

	Example
	* Create an object corresponding to a Customer record

* Assumes the record pointer is on the relevant customer

SCATTER NAME oCust MEMO

* Now you can refer to the fields as properties of the object

? oCust.Company_Name

? oCust.Customer_Id

* After editing, you can:

GATHER NAME oCust


	See Also
	Append From Array, Copy To Array, CursorSetProp()


_Screen

This system variable makes it easy to manipulate the main Visual FoxPro window. It contains an object reference to the window that you can use to manipulate the window's properties. 

	Usage
	_SCREEN.Property = uValue

uValue = _SCREEN.Property

_SCREEN.Method()


The screen object is a little strange. It contains some properties that apply to forms (like Controls and ControlCount) and others that apply to formsets (like Forms, FormCount and ActiveForm). It doesn't respond to any events, but it does have a small set of methods. The SaveAsClass method lets you create a visual class with the same characteristics as the window. However, it loses the formset properties en route and becomes a form-based class. 

Some of the things you can do with _SCREEN, you can also do with MODIFY WINDOW. But it's much more OOP-y to change properties using _SCREEN, and far more (though not all) of them are available. However, after you've made a whole lot of changes, it's hard to remember how you found things. No matter how you made the changes, you can restore the main window to just about the condition you found it in by issuing MODIFY WINDOW SCREEN with no arguments. 

	There are a few items that aren't restored by this—one that we know of is DrawWidth. Some items are restored in practice, but _SCREEN's properties are wrong. BackColor is one like that.



One of the most common uses we find for _SCREEN is getting access to the currently active form through the ActiveForm property. This property contains an object reference to the form that has focus. Since toolbars are never ActiveForm, they can use _SCREEN.ActiveForm to call methods of the current form. _SCREEN.ActiveForm is also handy in the debugger to figure out what's going on. 

Don't confuse _SCREEN with _VFP. That variable gives you access to the VFP Application object, an Automation server with a whole bunch of PEMs of its own. While we primarily used _SCREEN as the "top" object in VFP 3.0 _SCREEN-based systems, with the introduction of top-level forms and Automation servers, we're more likely to use the Application object. 

Beginning with VFP 7, the distinction between _SCREEN and _VFP becomes even larger. In older versions, the two objects shared those properties that describe the physical location of VFP on the monitor screen: Left, Top, Height and Width. Now, each has its own values for those items. _VFP's properties describe the entire space occupied by VFP; _SCREEN's properties describe the inside client area, omitting the area occupied by such things as the title bar, menu, status bar, and any docked toolbars. Having both sets of information available makes it easier to perform tasks that require precise positioning of forms or other objects. (Interestingly, when VFP is maximized, _VFP.Height and _VFP.Width are a few pixels larger than the available screen resolution. Our take is that maximizing puts the application's borders just off-screen.) 

_SCREEN gained several other properties in VFP 7. hWnd provides a window handle to VFP's client area that makes using a number of Windows API functions easier. _VFP also has the hWnd property; as with the positional properties, it reflects the main VFP window. 

_SCREEN also has the ShowInTaskBar property, but it's read-only here, making it pretty much useless. 

	Example
	* Personalize the screen

_SCREEN.Caption = "Hacker's Visual FoxPro"

_SCREEN.BackColor = RGB(0,0,255)

_SCREEN.ForeColor = RGB(255,255,255)

* or:

MODIFY WINDOW SCREEN TITLE " Hacker's Visual FoxPro" COLOR W+/B

 

* Call a method of the active form without knowing what form 

* is in charge

_SCREEN.ActiveForm.SaveRecord()


	See Also
	ActiveForm, hWnd, Modify Window, SaveAsClass, ShowInTaskBar, _VFP


Scroll

Don't confuse the useful Scrolled event with this legacy command. While it's a neat command, we don't expect anyone to use it in VFP. 

	Usage
	SCROLL nTopRow, nLeftCol, nBottRow, 

       nRightCol, nRowChange [, nColChange ]


	Parameter
	Value
	Meaning

	nTopRow
	Numeric
	Top row of area to be scrolled.

	nLeftCol
	Numeric
	Leftmost column of area to be scrolled.

	nBottRow
	Numeric
	Bottom row of scrolled area.

	nRightCol
	Numeric
	Rightmost column of area to be scrolled. 

	nRowChange
	Numeric
	Number of rows to move either up (negative) or down (positive).

	nColChange
	Numeric
	Number of columns to scroll area either right (positive) or left (negative).



Text and images on the screen or active output window can be moved around using this command. Areas they leave behind are erased. Text and graphics that scroll beyond the specified boundaries are lost and cannot be retrieved by scrolling in the opposite direction. 

	Specifying an nRowChange value of zero with no nColChange value (or a value of zero) erases the area within the specified row and column boundaries. However, an nColChange setting of zero scrolls the area only vertically. We feel that specifying changes of zero ("no change") should result in, well, no change.



Visual FoxPro translates the rows and columns specified into those based on the current output font; funny effects can result from screens that may have looked okay in DOS. This command was developed long ago for DOS-based products; we're surprised to find it works at all. 

	Example
	CLEAR

@ 5,5 SAY "C:\WINDOWS\WINLOGO.BMP" BITMAP

@ 10, 10 SAY "Way Cool Effects" ;

         FONT "Times",16 STYLE "T" COLOR R+

SCROLL 0,0,20,15,2

SCROLL 5,16,20,21,-2

FOR i = 4 TO -8 STEP -1

  IF i # 0 

    SCROLL 0,26-i,20,26-i,i*.1

  ENDIF

NEXT


	See Also
	@...Say


_Shell

_SHELL can be used to replace the command window with a different program shell. 

	Usage
	_SHELL = cCommand

cCommand = _SHELL


Some users, we've observed, are just not safe with the interactive development version of Visual FoxPro. (Some of them aren't safe with computers at all, but that's another topic...) To think about one of these folks getting access to the Command Window makes us wonder why we didn't get into some safe profession, like bungee testing. If the user absolutely, positively has to have the development version of Visual FoxPro, you can use the _SHELL variable to always run your program instead of dropping back to the Command Window. You can set this in your configuration file with the equivalent _SHELL command. 

Each time _SHELL is invoked, it clears the _SHELL setting, so you'll need to reset it in your application. 

If you think your users are absolutely, completely safe with your application while they run the interactive version, think about these three little letters: Z...A...P... 

	Example
	_SHELL = "DO SAFEMENU.PRG"


	See Also
	_Assist, Run


Spinner

A spinner is a gadget dedicated to entering numbers. It lets you enter them directly as well as increment or decrement the current value with either the mouse or the keyboard. FoxPro 2.x had spinners, but they were really just fancy text boxes. These spinners are much better because they give you tremendous control over what's going on. 

	Property
	Value
	Purpose

	Increment
	Numeric
	Determines the amount the value changes with a click of an arrow or pressing an arrow key.

	KeyboardHighValue, KeyboardLowValue
	Numeric
	Determine the highest and lowest numbers that can be entered from the keyboard.

	SpecialEffect
	Numeric
	Determines the appearance of the spinner: 3-D (0); Plain (1), meaning flat; or Hot Tracking (2), meaning flat until the mouse is over it. 

	SpinnerHighValue, SpinnerLowValue
	Numeric
	Determine the highest and lowest numbers that can be entered by "spinning."

	Text
	Character
	Contains the spinner value as an unformatted character string.


	Event
	Purpose

	DownClick, UpClick
	Occur when the down-arrow or up-arrow is clicked or the keyboard arrows are used.

	MouseEnter
	Fires when the mouse moves into the area occupied by the spinner.

	MouseLeave
	Fires when the mouse moves out of the area occupied by the spinner.

	RangeLow, RangeHigh
	Occur when the spinner tries to lose focus. Each can return a value, and the spinner's value must be between those two values for it to lose focus.

	Unlike some other controls, a spinner with SpecialEffect set to Hot Tracking does not take on the 3-D appearance. Since there's no documentation for this, and text boxes and edit boxes behave the same way, we're generously attributing this one to design. It's most noticeable when you move the mouse over the spinner, so it goes 3-D, then click into it, then move the mouse away so you can see what you're typing.



Check out Increment for an example of a spinner whose increment changes as the value changes. 

	Example
	* Set up a spinner class that counts

* by 100's between 1000 and 20000

* and annoyingly beeps when you use the spinner arrows.

oForm = CREATEOBJECT("Form")

oForm.AddObject("spnBigSpin", "BigSpinner")

oForm.spnBigSpin.Visible = .T.

oForm.Show()

 

DEFINE CLASS BigSpinner AS Spinner

 

   Value = 1000

   Increment = 100

   KeyboardLowValue = 1000 

   SpinnerLowValue = 1000

   KeyboardHighValue = 20000

   SpinnerHighValue = 20000

 

   PROCEDURE DownClick

   ?? CHR(7)

   ENDPROC

 

   PROCEDURE UpClick

   ?? CHR(7)

   ENDPROC

ENDDEFINE


	See Also
	DownClick, Increment, KeyboardHighValue, KeyboardLowValue, MouseEnter, MouseLeave, RangeHigh, RangeLow, SpecialEffect, SpinnerHighValue, SpinnerLowValue, Text, UpClick


_StartUp

This system variable lets you specify a program to run when FoxPro starts up. You specify it in the CONFIG.FPW file. 

	Usage
	_STARTUP = cProgram


This variable was added in FoxPro 2.6 to enable the Catalog Manager to be up and running when FoxPro opened. The specified program is run before a command specified by COMMAND=. 

Changes to _STARTUP once FoxPro is running don't have any automatic effects. But because this is a system variable that can't be cleared, it's another hook you can grab to use for your own purposes. See _ASSIST for ideas about how to use these hooks. 

	Example
	* The next line would appear in CONFIG.FPW.

_STARTUP = MyStart.PRG


	See Also
	_Assist, Configuration Files, _Shell


StatusBar, StatusBarText

These properties control the contents of the status bar. StatusBar is an application property that sends a message to the status bar. StatusBarText lets you specify some text to appear when a control has focus. It's a cousin to ToolTipText, but more useful for overall data entry. Use StatusBarText to give the user more information about what needs to be entered than the control or its associated label holds. Think of this as the first level of online Help. Like the messages associated with menu items, StatusBarText is there all the time, but is so unobtrusive that it won't bother the most experienced user. 

	Usage
	oApp.StatusBar = cHelpText

cHelpText = oApp.StatusBar

oObject.StatusBarText = cHelpText

cHelpText = oObject.StatusBarText


StatusBar entered the language in VFP 5 as a property of the application object. When you change its value, the new value is displayed on the status bar. But it's erratic. Lots of things clear it and restore the default value. We haven't been able to find any rhyme or reason to what changes the status bar and what doesn't. (In fact, we think the VFP developers have some trouble with this, too, because we've seen some pretty strange behavior in development mode. Without doing anything she knows of to cause it and without the mouse positioned over the standard toolbar, Tamar often sees the string "Standard" in the status bar for long periods of time. At other times, the StatusBarText from one of the buttons on that toolbar sticks around, even after the mouse is moved. In both cases, issuing SET STATUSBAR ON reactivates the status bar.) 

StatusBarText is a property of individual controls. There's a gotcha when you enter StatusBarText (and other strings) through the Property Sheet. Don't surround the text with quotes (yeah, we know it feels unnatural)—if you do, the quotes show up when you run the form. On the other hand, when you assign StatusBarText in code, you do need the quotes. It makes sense (because the Property Sheet, like the Query Designer, is smart enough to figure out what type of data it wants and deal with it appropriately), but it's really disconcerting with a property as text-oriented as this one. 

Grids have StatusBarText, but it appears only when the grid first receives focus and only if you land on a control in the grid that doesn't have its own StatusBarText. As you move around in the grid, the grid's StatusBarText does not reappear. You see it again only when focus moves to another control and then back to the grid. We'd like the grid's StatusBarText to appear anytime focus is on a column whose control doesn't have any StatusBarText. That would be truly useful. As is, a grid's StatusBarText seems pretty useless. 

	Example
	* For a Save button, you might set:

This.StatusBarText = " Save the Current Record"


	See Also
	ErrorMessage, Message, ToolTipText


TabIndex

This property determines the tab order of controls within a container. It's maintained separately for different containers, so you can do things like order the buttons in a group separately from the order of controls on the form. 

	Usage
	oObject.TabIndex = nTabOrder

nTabOrder = oObject.TabIndex


This property determines the sequence of data entry in your form—the route through the controls the user will take if she presses the arrow keys or Tab to move from one control to the next. The order in which a user tabs through the controls in a form does not have to be the same as the creation order of the controls. TabIndex lets you determine the tab order. Every container except grids and page frames, which have different properties for this (so much for polymorphism), lets you set TabIndex for each control inside that accepts focus. When you tab (or arrow, in some cases) through the controls in that container, the order you see is the TabIndex order of those controls. 

The structure used here is well thought out. Rather than numbering every control on a form in one sequence at the form level, you specify the order only for those controls directly contained on the form. If some of those controls contain others (for example, command groups contain command buttons), you number the items inside that container independently from the form. So each command group, option group and page has its own numbering sequence, no matter how deeply it's nested. The container has a TabIndex (or in the case of pages, PageOrder) that determines its sequence within the form or page that holds the container. This approach makes it easy to reorder things without fouling up the work you've already done. 

At design-time, the Tab Order option on the View menu lets you manipulate TabIndex without having to fiddle in the Property Sheet. Think of it as an "Order Builder." 

We suggest you use the Tab Order menu option, rather than manipulating values on the Property Sheet. There are two methods of ordering available at design-time. You choose the one you want in the Tools-Options dialog, under the Forms tab. You can set Tab Ordering to be either "Interactive" or "By List," our preference. It's easy to get confused about the TabIndexes of objects in a group. When you change one, the others don't change to fill in the gaps, so you can end up with several objects with identical TabIndexes. (Don't know why VFP can't adjust them. Other properties like PageOrder are smart enough to do this. On the other hand, that automatic adjustment makes manipulating PageOrder really tough, so maybe there's no good solution to this one.) 

	You can use the Tab Order dialog to reorder at any level. Just right-click, Edit on the container you want to reorder, then choose View-Tab Order from the menu. The dialog lets you reorder the objects inside the chosen container. Very cool!



You can change the TabIndex at runtime, too (though it suffers from the same confusion as at design-time), but we suggest you don't. Except in very unusual circumstances, changing the order in which the user tabs through the controls definitely falls into the category of a user-hostile interface. 

	Example
	ThisForm.cmgNavButtons.cmdTop.TabIndex = 1


	See Also
	ColumnOrder, CommandGroup, OptionGroup, Page, PageOrder, ZOrder


TabStop

This property determines whether you can get to a control with the keyboard. 

	Usage
	oObject.TabStop = lCanTabHere

lCanTabHere = oObject.TabStop


You can use TabStop to make some items mouse-only. For example, you can emulate the Mac interface by setting TabStop to .F. for all buttons, check boxes and the like. (We can't imagine why anybody would want to work that way, but it is the Mac standard.) If your users demand this interface convention, make sure all of your graphical controls have hot keys, so reaching for the mouse isn't mandatory to make it through a form. 

TabStop is most useful when you want, for example, to let the user tab through the data-entry fields without having to tab through all the buttons, too. The downside of this, of course, is that the user then has to resort to button hot keys or switch to the mouse to save her data. 

You can change TabStop at runtime, but we don't recommend it. It seems like a good way to terminally confuse your users. 

	TabStop is ignored for controls in a column of a grid. The docs say that it's read-only in that case, but in fact you can change it. Your changes are ignored, however. Seems that controls in a grid, if available, must be tab-able.


	Example
	* Keep user from tabbing into all buttons on a form

* Actually this only affects buttons based on the base classes

ThisForm.SetAll("TabStop", .F., "CommandButton")

ThisForm.SetAll("TabStop", .F., "OptionButton")


	See Also
	Enabled, ReadOnly, TabIndex


_Tally

This system variable tells you how many records were processed by the last command that affects it. The most common use for _TALLY is to check how many records were returned by a SQL-SELECT, but it actually works with many Xbase commands and the other SQL commands. 

	Usage
	nNumAffected = _TALLY


Here is the complete list of commands that change _TALLY (well, we think it's complete—we tried a bunch of undocumented stuff): 

	Append From
Average
Blank
Calculate
Copy To
Copy To Array
Count
	Delete
Delete-SQL
Export
Index
Pack
Recall
Reindex
	Replace
Replace From Array
Select-SQL
Sort
Sum
Update-SQL



Actually, JOIN, TOTAL and UPDATE are documented as changing _TALLY, too, but they're so obsolete, we didn't even test them. A good thumbrule is that _TALLY is updated by every command that displays a record count when TALK is ON. 

Generally, _TALLY contains the number of records affected by the command. So, after DELETE, it has the number of records deleted; after REPLACE, _TALLY contains the number of records that had a field replaced. 

	However, there's one real oddity on this list. _TALLY after REINDEX contains the number of records for which the last tag indexed applies. If you have filtered tags (indexes using a FOR clause), _TALLY returns the number of records contained in the last index. "Last" here refers to the creation order of the tags, since they're re-created in the same order. This isn't really a big problem since we suggest you never use REINDEX anyway.

	In VFP 3 and VFP 5, when TALK is OFF, INDEX and REINDEX don't update _TALLY. This bug is fixed in VFP 6.


	Example
	SELECT First_Name, Last_Name FROM Employee ;

   WHERE MONTH(Birth_Date)=MONTH(DATE()) ;

   INTO ARRAY aThisMonth

* Check if any records met the criteria.

IF _TALLY=0

   * If not, create the array with blank values.

   DIMENSION aThisMonth[1,2]

   aThisMonth = ""

ENDIF


This example shows perhaps the most common use of _TALLY. When you SELECT INTO an array and no records are selected, the array doesn't get created. So, it's usual to test _TALLY right afterwards and, if necessary, create the array. 

However, the wide range of commands that affect _TALLY lends itself to all kinds of possibilities. We bet most of you have sections of code that perform one of the commands above, then use COUNT or RECCOUNT() or ALEN() to see how many records were affected (we sure do). And there was _TALLY just waiting to give you the answer. Here's a simple example: 

USE Employee

COUNT FOR MONTH(Birth_Date)=MONTH(DATE())

WAIT WINDOW ;

     LTRIM(STR(_TALLY))+" employees have birthdays this month"

Do watch out for one thing with _TALLY. Because so many commands affect it, it's important to grab the value right away if you want to keep using it. Otherwise, you run the risk that a later command will overwrite the value. 

	See Also
	Append From, Average, Blank, Calculate, Copy To, Copy To Array, Count, Delete, Delete-SQL, Export, Index, Join, Pack, Recall, Reindex, Replace, Replace From Array, Select-SQL, Sort, Sum, Total, Update, Update-SQL


_TaskList, _FoxTask

These system variables, new in VFP 7, reference the Task List manager application and the Task List table, respectively. 

	Usage
	_TASKLIST = cTaskListApplication

cTaskListApplication = _TASKLIST

_FOXTASK = cTaskTable

cTaskTable = _FOXTASK


	Parameter
	Value
	Meaning

	cTaskListApplication
	Character
	The filename, including path, of the application that manages the task list. By default, HOME() + "TaskList.APP".

	cTaskTable
	Character
	The filename, including path, of the table containing tasks. By default, "FoxTask.DBF" in the user's Visual FoxPro profile directory. (In Windows 2000, "C:\Documents and Settings\<user name>\Application Data\Microsoft\Visual FoxPro".)



The task list provides an easy mechanism for managing a "to do" list in VFP. You can add items, called shortcuts, to it from any code-editing window. The Task List manager application, accessed by choosing Tools | Task List from the menu, provides an interface for the data. 

As with so many other components of VFP, the Task List manager application is written in VFP. The _TASKLIST variable allows you to replace the application provided with one of your own. The source code is provided as part of the XSource.ZIP file, installed by default in the HOME()+"Tools\XSource" directory. 

You can run the Task List application by issuing DO (_TASKLIST) or by selecting Tools | Task List from the development menu. As far as we can tell, you can even include the Task List manager in your own applications; we can't find any evidence to suggest that it's among the restricted files. However, the initial version of the tool is somewhat quirky and, of course, your users can't use VFP's built-in shortcut functionality. On the whole, you might be better off using the code as a guide to creating your own task list, better suited to your application's needs. Even better, you can use Automation to control Outlook or another application that already has task-list functionality. 

_FOXTASK lets you store the task list data where you prefer, rather than the default location. The table you specify must exist before you can change _FOXTASK. It also must contain at least one record, with the first record containing version data. The easiest way to create such a table is to copy the default task list table, and get rid of any extraneous records. 

If _FOXTASK is empty when you start the Task List application, a new task list table is created in the VFP home directory (and named FoxTask.DBF). 

	Example
	_FOXTASK = HOME() + "Tasks.DBF"

DO (_TASKLIST)


	See Also
	_Browser, _Builder, _Coverage, _Gallery, _Wizard


TerminateRead

This is another property included only for conversion of FoxPro 2.x screen sets. It corresponds to the Terminate Read option offered by several 2.x controls. When this property is .T. for a control, using that control shuts down the formset. 

	Usage
	oObject.TerminateRead = lEndItAll

lEndItAll = oObject.TerminateRead


This property is relevant only for formsets running in one of the READ compatibility modes (with WindowType = 2 or 3). It's ignored otherwise. 

	See Also
	Release Method, WindowType


Value, Text

Value is a wonderfully overloaded property that, in one way or another, tells you the current value of a control. The type and interpretation of the value changes, depending on the control involved. Text is a handy corollary to Value, giving you a textual version of the value, in case it's not character. Text applies only to the controls that allow textual input. 

	Usage
	oObject.Value = uControlValue

uControlValue = oObject.Value

cValueText = oObject.Text


You can change the Value of a control and see your changes reflected in the display. In some cases, changing the Value can have profound effects. In at least one, it can cause ugly behavior. 

The simplest cases are text boxes, edit boxes and spinners. In that case, what you see is what you get. Value contains the actual value displayed. 

Except for a couple of things. If an edit box has MaxLength set so you can only see part of the data, Value still contains the entire string. Watch out, though: as soon as you start fiddling with the data in the edit box, the data is truncated to MaxLength characters. If the edit box is bound to a field, the data in the field changes, too. 

Value's type for these controls is determined by the control's ControlSource, if there is one. If not, Value is character for text and edit boxes and numeric for spinners. 

These controls plus combo boxes are the ones that have the Text property. Most of the time, Text contains what you'd get if you applied TRANSFORM() to Value; that is, a character string that looks like the value. However, some of the Format options change that. For example, when Format includes "R", Text includes the extra formatting characters, while Value does not. When Format includes "E" (for British dates), Text contains the date in the DD/MM/YY format rather than the current date format. 

On to buttons. Command buttons don't have a Value. Only CommandGroups do. The value can be either numeric or character. If it's numeric, it contains the position in the group of the button last chosen. (Position is based on the buttons' order in the Buttons collection.) If Value is character, it contains the Caption of the last chosen button. 

OptionGroups work the same way, allowing either numeric or character Values. However, individual option buttons also have a Value property. It's 1 if that button is chosen and 0 if it isn't. 

	You can set the Value of more than one option button in an option group to 1, so that multiple options appear to be chosen. You can also set them all to 0, so that no buttons are chosen. Individual option buttons shouldn't even have a Value property, but since they do, it ought to work right.



Check boxes are pretty simple, too. Their Value can be numeric or logical, with 0 = .F. = unchecked and 1 = .T. = checked. There's one complication: A check box can also show a null value. Set Value to 2 or .NULL. and the check box turns gray. Watch out. Just because it's gray doesn't mean it can't be clicked—it can. Use Enabled to make it unclickable. 

For combo and list boxes, things get interesting. Value can be either character or numeric. If it's character, it contains the actual text of the chosen item—as a character string. Even if the RowSource data is something other than character, when it's placed in a list or combo, it's converted to character there. If Value is numeric, it's the Index in the list or combo of the chosen item. In VFP 5 and later, you can put a numeric data item in Value (as a character string, of course) by setting the list or combo's BoundTo property to .T. This lets you show a description in the list while saving a numeric ID for the Value. In VFP 3, there's no way to use numeric data for the Value of a list or combo. 

For multi-column lists and combos, Value (when character) contains the contents of BoundColumn for the chosen item. DisplayValue, however, always contains the contents of the first column, unless you make it numeric. 

Finally, the one that surprised the heck out of us: Grids have Values. The Value is available only when the grid has focus, but then it contains the value of the cell with focus. The type of Value changes to match the current column. When the grid doesn't have focus, its Value is 0. 

	Don't change a grid's Value—doing so unlinks the connection between the grid contents and the Value property. Once you do it, Value is no longer updated by moving in the grid.



Beware. When a control has a ControlSource, the new Value doesn't reach the ControlSource until just before the control's Valid. If you check the value of the ControlSource in InteractiveChange, you won't see the new value. That is, in InteractiveChange, Value shows one thing and the value of the ControlSource shows something else. Yet another reason to always refer to Value, not the actual ControlSource to which the control is bound. 

	Example
	* Initialize a list to a particular item.

This.Value = cCurrentChoice


	See Also
	BoundColumn, BoundTo, Buttons, Caption, ControlSource, DisplayValue, Enabled, Format, InteractiveChange


Join, Total, Update

As long as we've been using Xbase, we've heard that these three commands should be avoided. At least as far back as FoxBase+, the manuals warn that JOIN may be very slow and can overrun available disk space. 

As a result of these dire warnings, we never looked very hard at these commands (though we've occasionally been called on to maintain code that uses them). Now that we've done so, we've decided it's just as well. They're hard to use, limited in capability, and can be extremely slow. 

	Usage
	JOIN WITH cAlias | nWorkArea TO cTable 

        FOR lForExpression 

        [ FIELDS cFieldList ]

        NOOPTIMIZE


JOIN is sort of a limited version of SELECT-SQL. It matches up records in a pair of tables based on some condition and creates a new table (cTable) containing fields from both the original tables. One table must be open in the current work area; the other must be open in another work area. The FOR clause determines how the records are joined. You can limit the fields in the result using the FIELDS clause, but JOIN can't handle memo or general fields, and you can't use expressions in the field list. 

For very small data sets, JOIN can be competitive in speed with SELECT, but as soon as you get beyond a handful of records, it bogs down. The thought of throwing NOOPTIMIZE into the mix is really frightening. 

	JOIN has a bug when dealing with long field names. In the newly created table, all field names are truncated to 10 characters or less. This makes some sense since the table is created as a free table. 



Frankly, we can't see any reason to use JOIN ever. Anything you can do with it, you can do more easily with SELECT-SQL. Even Microsoft thinks so, as they've labeled this command "for backward compatibility," and suggest you use SELECT-SQL. 

	Example
	* Here's an example involving very small tables so you

* can see the result in this lifetime. It fails in 

* versions VFP 5 and later, but does work in VFP 3.

USE category

SELECT 0

USE products

JOIN WITH category TO temp FOR category_id=category.category_id

 

* Here's the SELECT-SQL equivalent:

SELECT * FROM category, products ;

   WHERE category.category_id=products.category_id ;

   INTO TABLE temp


	Usage
	TOTAL TO cTable ON uField

        [ FIELDS nNumericFieldList ]

        [ Scope ]

        [ FOR lForExpression ]

        [ WHILE lWhileExpression ]

        [ NOOPTIMIZE ]


TOTAL is similar to working with GROUP BY in SELECT-SQL, but far more limited. It groups records from the table open in the current work area based on the listed field (uField) and totals the numeric fields listed in nNumericFieldList. If FIELDS is omitted, all numeric fields are totaled. A new table is created to hold the result. All fields from the original table are included in the result. You can limit the records included in the computation using the Scope, FOR and WHILE clauses. 

So what else is wrong with TOTAL? The new table created has the same structure as the original. If the total for a field is too big to fit, you lose precision. Also, since all fields are carried along, you get random results for fields that aren't either totaled or listed in the ON clause. (Actually, this is the same thing that happens when you include extraneous fields in a GROUPed query, but at least there you have a chance to omit them.) TOTAL is limited to working with a single table—you can't match up records from multiple tables. Finally, you can't compute expressions, only sum existing fields, so you can't do things like total (quantity*unit price) to get an invoice total. 

Bottom line: Like JOIN, we can't see a lot of reasons to use TOTAL. 

	Example
	* We couldn't find a way to use TOTAL with the sample

* TasTrade database that comes with Visual FoxPro.

* The table in this example is from the Tutorial 

* data that shipped with earlier versions of FoxPro.

USE Detail       && Invoice detail lines

* Compute invoice totals

TOTAL ON ino TO InvTotal FIELDS Ltotal


	Usage
	UPDATE ON uJoinFieldName FROM cTable

        REPLACE uFieldName1 WITH eExpression1

        [ , uFieldName2 WITH eExpression2 [ , ... ] ]

        [ RANDOM ]


UPDATE is a sort of early way of creating a relation between two tables and using it to modify data. The table in the current work area gets updated based on establishing a relation between it and another open table. The relation is based only on a single identically named field (uJoinFieldName) in the two tables. The current table must be ordered on the common field, either with an index or the natural order of the records. The other table (cTable) should also be ordered on the common field. If not, you must include the RANDOM keyword. 

For each matching record found in cTable, each listed field is updated as specified in the REPLACE clause. The expression eExpressionn can reference fields from both tables (actually, we suppose, from any open tables). 

Got all that? It took us a while. This is an incredibly complicated command to do something that there are much easier ways to do. UPDATE-SQL lets you do similar things a lot more simply since you can specify the relationship explicitly in its WHERE clause. But, even in FoxPro 2.x, you could do this sort of thing by setting a relation and then using REPLACE. 

In addition to the difficulty of using it, in a properly normalized database, there aren't likely to be a lot of places where you'd want to do this sort of computation. The only thing we can think of is updating a table from a copy, and that's often easier in Visual FoxPro just by using the built-in buffering techniques. 

We just can't see any reason to use UPDATE. 

	Example
	* Again, we can't see a way to use UPDATE with

* the nicely normalized TasTrade database.

* The example uses the old, less normalized, 

* Tutorial data.

SELECT 0

USE Detail

SELECT 0

USE Invoices

 

* Clear out existing invoice totals.

REPLACE ALL iTotal WITH 0

 

* Now recompute.

UPDATE ON ino FROM Detail ;

   REPLACE iTotal WITH iTotal+Detail.Qty*Detail.Price


	See Also
	Replace, Select-SQL, Update-SQL


UIEnable

This event gives you a hook into the controls on a page when the page is activated or deactivated. It fires for each control on the page on the way into and on the way out of the page. It took us awhile to "get" this one, but it's actually pretty cool. 

	Usage
	PROCEDURE oObject.UIEnable

LPARAMETERS [ nControlIndex ,] lComingIn


	Parameter
	Value
	Meaning

	nControlIndex
	Numeric
	If the control is contained in an array, indicates which element of the array fired the event.

	
	Omitted
	The control is not part of a control array.

	lComingIn
	.T.
	The event fired because the page containing this control is being activated.

	
	.F.
	The event fired because the page containing this control is being deactivated.



The firing of this event is tricky. First, it fires not for the page frame, nor for the page (in fact, pages don't even have a UIEnable event), but for the controls contained on the page becoming active or inactive. That is, when a page comes to the top, the UIEnable events of the controls for the page that was on top fire, then the UIEnable events for the controls on the page that's now on top fire. 

In addition, UIEnable fires only on changes to the active page. If focus leaves the page frame, it doesn't fire. Nor does it fire when focus returns to the page frame. It fires only when a different page of the page frame comes to the top. 

When would you use this event? When you want to check something about some control whenever its page comes to the top. If a control on one page depends on several other controls in the page frame, UIEnable lets you update it once—when you need it to be right. For example, suppose a list is based on a view and the view parameters are specified on the other pages. You don't want to requery the view and the list every time one of the parameters changes—that would be horribly slow. Instead, you can put the calls to the REQUERY() function to update the view and the Requery method to update the list in the UIEnable method of the list and it'll be updated whenever that page comes to the top. 

You might also take advantage of this event to refresh pages when they come to the top. Many folks add a custom object to each page in a page frame with the UIEnable method set to call the page's Refresh method on the way in. (The adding is done in the Init method.) 

This event is really elegant. On the other hand, we can't figure where its name came from. Probably from some other Microsoft language. We think PageActivate and PageDeactivate (or OnPageActivate and OnPageDeactivate) would be just a wee bit more intuitive. 

	Example
	* In a simplified version of the example above,

* assume one page of a page frame contains a list

* based on a query (RowSourceType=3) and that

* the query conditions include some variables,

* which are the ControlSources of controls

* on other pages.

* To update the list when its page comes to

* the top, you'd have the following in

* the list's UIEnable event:

IF lEnable   && Only on the way in

  This.Requery()

ENDIF


	See Also
	Page, PageFrame, Refresh


VFPXMLProgID

This property of the _VFP object, new to VFP 7, allows you to specify a COM component to override the default functionality of VFP's internal XML functions: CursorToXML(), XMLToCursor(), and XMLUpdateGram(). If you don't like the way these functions work, you can roll your own and put them in your own COM object. 

	Usage
	CProgID = _VFP.VFPXMLProgID

_VFP.VFPXMLProgID = cProgID


Initially, you can query _VFP.VFPXMLProgID and it returns an empty string (""). Setting this property to the empty string tells VFP to use the internal VFP functions. 

When you set the property to the ProgID of a specially designed COM object, VFP uses the methods in the COM object instead of the internal functions. 

So how do you define this COM object? It must implement Visual FoxPro's IVFPXML interface. You can write it in Visual C++, Visual Basic, Visual FoxPro, or any other language that can create COM objects with the ability to implement interfaces. (For more on implementing interfaces in VFP, see "Face to Face with COM Interfaces" in "It Was Automation, You Know.") In VFP, the easiest way to do this is to open the Object Browser, then open the Microsoft Visual FoxPro 7.0 type library. Scroll down and expand the Interfaces node, then drag the IVFPXML icon to a new program. You'll get something like: 

x=NEWOBJECT("myclass")

 

DEFINE CLASS myclass AS session OLEPUBLIC

 

  IMPLEMENTS IVFPXML IN ;

           "c:\program files\microsoft visual foxpro 7\vfp7.exe"

 

  PROCEDURE IVFPXML_CursorToXML( ;

      bstrAlias AS STRING, nOutputFormat AS Number, nFlags AS Number, ; 

      nRecords AS Number, bstrOutputFile AS STRING, bstrSchema AS STRING, ;

      bstrSchemaLocation AS STRING, bstrNameSpace AS STRING, ;

      pVFP AS VARIANT) AS VARIANT;

    HELPSTRING "Converts from a Cursor to XML"

    * add user code here

  ENDPROC

 

  PROCEDURE IVFPXML_XMLToCursor( ;

      pvarXMLSource AS VARIANT, bstrCursorName AS STRING, ;

      nFlags AS Number, pVFP AS VARIANT) AS Number;

      HELPSTRING "Converts from XML to a Cursor"

    * add user code here

  ENDPROC

 

  PROCEDURE IVFPXML_XMLUpdateGram(;

      nFlags AS Number, bstrCursorList AS STRING, ;

      pVFP AS VARIANT) AS VARIANT;

      HELPSTRING "Generates an XML UpdateGram"

    * add user code here

  ENDPROC

 

ENDDEFINE

At this point, you're on your own. You're free to develop whatever replacement you'd like for CursorToXML(), XMLToCursor(), and XMLUpdateGram(). So if the built-in functions don't do things the way you want, or you've got some customized routines that are written in another language that you'd like to use, the VFPXMLProgID property lets you choose an alternative. 

	Example
	* Assumption: You have a COM object that does the conversions,

* and it's based on the above code, and has the ProgID of

* "AlternateXML.MyClass".

 

_VFP.VFPXMLProgId = "AlternateXML.MyClass"


	See Also
	CursorToXML(), Define Class, _VFP, XMLToCursor(), XMLUpdateGram()


Wait

WAIT is a simple little command, but with nice effect. Its primary mission in life is to display a message. By default, a WAIT WINDOW command appears in the upper right corner of the active screen, below the toolbars, while a plain WAIT is displayed in the current output window. 

	Usage
	WAIT [ CLEAR ] | [ cMessage ][ TO cResponse ]

[ WINDOW [ AT nRow, nColumn ] [ NOWAIT ] [ NOCLEAR ] ]

[ TIMEOUT nTime ]


	Parameter
	Value
	Meaning

	CLEAR
	 
	Clears any existing WAIT WINDOW. Cannot be used with any other keywords or arguments!

	cMessage
	Character 
	The message to be displayed. Multiple lines can be specified by separating strings with carriage returns (CHR(13)'s). More than 254 characters generates error 1903, "String too long to fit."

	
	Omitted
	Display "Press any key to continue..."

	
	Empty String
	Show the cursor, blinking, at the last output location, unless you have SET CURSOR OFF.

	cResponse
	Character 
	Receives a one-character alphanumeric response. Blank is returned for nonprintable characters or mouse clicks. The specified memory variable is created if it doesn't exist. If it does exist, its type is changed to character, if necessary, and its value is overwritten.

	nRow
	Numeric
	The row where the window should appear. The row number is based on the screen font (see WFONT()). Numbers less than 0 are ignored.

	nColumn
	Numeric
	The column where the window should appear. The column number is based on the screen font (see WFONT()). Columns less than 0 are ignored.

	NOWAIT
	 
	Continues processing immediately.

	NOCLEAR
	 
	WAIT WINDOW remains until another WAIT WINDOW command or WAIT CLEAR is issued. Without this clause, the window clears when a keystroke or mouse click is received.

	nTime
	Numeric
	The time to wait in seconds before continuing processing. Ignored if NOWAIT is specified.



Note that CLEAR and NOCLEAR are not a pair, as the Help would have you believe. CLEAR must be used alone to clear an existing window; any other use results in a syntax error. NOCLEAR is new to Visual FoxPro—it keeps a window on the screen even if a keystroke is pressed. 

The many clauses of WAIT that were added over the years can lead to some commands that look like they should work, and don't cause an error message, but don't give the expected response. For example, 

WAIT WINDOW NOWAIT "Processing..." TIMEOUT 30

might be interpreted as "Display this message, let me continue processing, and then clear the message after 30 seconds." In fact, the TIMEOUT clause is ignored if NOWAIT is specified, and the message clears only if a keystroke is pressed or a WAIT CLEAR command is issued. 

WAIT is getting dated. For input, consider instead a MessageBox or custom form. To keep the user informed of processing status, a custom form with the new ProgressBar control is more informative and more standard. Finally, look at the many options to update messages on the application status bar as well. WAIT WINDOWs give away that a FoxPro app is running—not necessarily a bad thing, but a non-standard interface that could confuse new users. 

	[image: image48]
	WAIT "" produces a blinking cursor in the screen or current output window. WAIT WINDOW "" does the same—where's the WINDOW? There ain't one.


	[image: image49]
	WAIT "" NOWAIT immediately executes the next command, with no display as you might expect. However, WAIT "xxx" NOWAIT leaves the cursor blinking on the "xxx" message and needs a key to return control to the program, ignoring the NOWAIT clause. In fact, even WAIT NOWAIT with no string waits. We hardly ever use NOWAITs without WINDOWs, because the window, particularly with the AT clause, is so cool. But if you're going to use it (consider using ? instead), or are porting code that uses it, don't expect it to NOWAIT just because you told it to.


	Example
	WAIT WINDOW "Sorry, Dave, I can't do that." NOWAIT NOCLEAR


	See Also
	Accept, Input, MessageBox(), SysMetric(), WFont()


WindowState

This property determines whether a form is minimized, maximized or shown at its normal size. 

	Usage
	frmForm.WindowState = nWindowState

nWindowState = frmForm.WindowState


	Parameter
	Value
	Meaning

	nWindowState
	0
	Normal—the current defined size.

	
	1
	Minimize to an icon.

	
	2
	Maximize to the size specified by MaxHeight and MaxWidth, or to fill the Visual FoxPro window.



You can achieve the same effects with ZOOM WINDOW, but changing WindowState is the OOP way to do this. 

You can use WindowState to minimize and maximize a form even if it doesn't have Min and Max buttons and even if the window can't be sized because of its BorderStyle. Changes to WindowState fire the form's Resize event, so you can take appropriate action. 

One weird behavior. If you maximize, then minimize a form, when you then set WindowState to 0 to restore the form, Resize fires twice because the form first maximizes, then returns to its original size. 

	Example
	* This code might be in a form's Resize event (if you're

* using formsets, which we don't)

IF This.WindowState = 2   && maximized

   * move things around so they can be seen

   ThisFormSet.Form2.Left = This.MaxLeft + This.Width

ENDIF


	See Also
	BorderStyle, MaxButton, MaxHeight, MaxWidth, MinButton, Zoom Window


WLast(), WOnTop(), WOutput()

These three functions tell you about active windows and where output is going. WONTOP() tells you which window is active now. WLAST() tells you which window was active before the one that's active now. WOUTPUT() tells you where output is going now. 

All three of these functions can either give you the name of the window that has the specified status (last active, currently active, receiving output) or tell you whether a particular window has that status. 

	Usage
	cReturnValue = WLAST()

lReturnValue = WLAST( cWindow )

cReturnValue = WONTOP() 

lReturnValue = WONTOP( cWindow )


WONTOP() without a parameter returns the name of the window that's "on top," that is, the active window. Normally, you can identify this window because its title bar is a different color than other windows' title bars. If you pass a window name, WONTOP() tells you whether or not it's on top. Toolbars can never be WONTOP(), though other system windows can. The Command Window is also never WONTOP(). 

If the parameter is omitted, WLAST() returns the name of the window that was active just before the current active window (the previous WONTOP()). If you pass the name of a window, the return value tells whether that window was active just before the current active window. 

When you click from one window to another, the window you click on becomes WONTOP(), while the one you left becomes WLAST(). Issuing ACTIVATE WINDOW makes the named window WONTOP(). However, the Command Window is never WONTOP() or WLAST(). 

There's a subtle point when using a form set (or a 2.x screen set). When you move from one form to another in the set, the DEACTIVATE method of the form you're leaving fires. By the time that happens, the form you're leaving has become WLAST() and the new form is WONTOP(). When using 2.x-style screen sets, the same thing is true for the DEACTIVATE clause of the READ command—by the time it executes, the new window is WONTOP() and the one you're leaving is WLAST(). 

However, except when building developer's tools, checking WONTOP() and WLAST() and changing them with ACTIVATE WINDOW should be much less common in Visual FoxPro than in FoxPro 2.x. Since pretty much every window involved in an application will be a form, you can do these manipulations using form and formset properties and methods instead. 

A hidden window can't be WONTOP() even if you use the optional SAME clause of HIDE WINDOW. A window must be visible to be WONTOP(). 

	Example
	* Open the Watch window

* window and put WONTOP() and WLAST() in it

* Then, try this:

DEFINE WINDOW test1 FROM 0,0 TO 10,30

DEFINE WINDOW test2 FROM 0,40 TO 10,70

ACTIVATE WINDOW test1

* notice that WONTOP() is now "TEST1"

* click into the Command window

* now WLAST() is "TEST1", too

ACTIVATE WINDOW test2

* WONTOP() becomes "TEST2"

* Click between the two windows and watch what happens.

* Note that clicking on a window doesn't make it WONTOP().

* This isn't true with forms.


	Usage
	cReturnValue = WOUTPUT()

lReturnValue = WOUTPUT( cWindow )


Without a parameter, WOUTPUT() tells you the name of the window to which any output will be sent (the "output window"). If you pass a window name, you find out whether that window is the output window. 

Issuing ACTIVATE WINDOW changes WOUTPUT() to the named window, if it's capable of receiving output. Clicking on a window created with DEFINE WINDOW does not make it WOUTPUT(), but clicking on a form does. 

WONTOP() and WOUTPUT() often have the same value, but not always. There are many windows that cannot be WOUTPUT() because they can't receive output. This is true of all system windows and of Browse windows. (In FoxPro 2.x, comparing WONTOP() to WOUTPUT() was one way of determining whether a Browse or a READ window was active when coordinating Browse with READ.) 

Output from commands (like DIR, ?, and COUNT) that you probably think of as going to the main FoxPro window actually goes to WOUTPUT(). To have this output go to the main FoxPro window, you can issue ACTIVATE SCREEN before the command that generates the output. However, as we've mentioned elsewhere, using the Screen raises various issues, because some applications may use only top-level windows and dispense with the Screen entirely. 

WOUTPUT() can point to a hidden window, which can be confusing when your output seems to disappear entirely. 

	Example
	* open the Watch window and put WOUTPUT() in it

* Then:

DEFINE WINDOW test1 FROM 0,0 TO 10,30

DEFINE WINDOW test2 FROM 0,40 TO 10,70

ACTIVATE WINDOW test1

* note that WOUTPUT() is now "TEST1"

ACTIVATE WINDOW test2

* WOUTPUT() is now "TEST2"

* Click on Debug Window - WOUTPUT() doesn't change

* Click between the windows - WOUTPUT() doesn't change


	See Also
	Activate Screen, Activate Window, ActiveForm, Define Window, Hide Window, Show, Show Window, WTitle()


_Wrap

This remnant of the printer variable system actually does something. It determines whether output produced by ? and ?? pays attention to _RMARGIN or not. If _WRAP is .T., output stops at _RMARGIN and wraps to the next line. 

	Usage
	_WRAP = lWrapIt


One other kink feels like a bug, but actually makes sense. When _WRAP is .T., output from ?? is stored up until either the right margin is reached or a ? command is issued. The characters aren't sent to the device until that time. Where this catches us is when we are trying to ring the bell with ?? CHR(07). Remember to turn _WRAP off before issuing that command. 

	Example
	_WRAP = .T.

_RMARGIN = 20

? "This long string will be wrapped when you see it"

_WRAP = .F.

? "This one won't be wrapped because _WRAP is .F."


	See Also
	_Alignment, _Indent, _LMargin, _RMargin


ZAP

This may be the single most dangerous command in all of FoxPro, but it's incredibly handy when you're working interactively. ZAP permanently removes all records from a table. The name supposedly stands for "Zero And Pack." ZAPped records cannot be recalled. 

	Usage
	ZAP [ IN cAlias | nWorkArea ]


Although you can ZAP in a work area other than the current one, we really don't recommend it. Using this command by itself is like striking matches; using it in another work area is like striking matches in a gas station. 

	ZAP is not a good database citizen. It doesn't call the Delete trigger for a table in a database. Instead, it neatly avoids all the work you've done to make sure the integrity of your database is maintained. This is a major flaw.



Due to the above and for lots of other reasons, never use ZAP in a program. It's just too risky. See PACK for suggestions on alternatives. Actually, the one case where ZAP is acceptable is when you're working on a temporary file in the first place, so no permanent data is at risk. In particular, it's a great way to re-create a cursor without losing the grid that's based on it. Simply ZAP and APPEND FROM your data instead of doing a SELECT. 

So why do we think it's incredibly handy? When you're manipulating data manually—perhaps parsing older data to create a normalized database—it's clean and simple to ZAP the target table between tests. Outside this kind of situation, we strongly recommend you avoid ZAP. 

ZAP does respect SET SAFETY, so if you have it on, you are warned before you throw all your data in the garbage can. 

	Example
	USE TestData

ZAP


	See Also
	Delete, Pack, Recall, Set Safety


